

Chapter 8

Printing Dot Graphics

In Chapter 7 you were introduced to a form of computer graphics; you were able to actually define characters dot by dot. In this chapter you'll learn to use the same principles to make Delta print whole pages of dot graphics! We'll show you how to use dot graphics to create "super download characters." In addition, you'll see how your Delta printer can be used as a graphics plotter. This can have some practical business applications as well as create some terrific computer art!

Comparing Dot Graphics with Download Characters

A good understanding of dot graphics requires an understanding of how dot matrix printers work; you may want to

review the first few pages of Chapter 7. The principles for dot graphics are the same as those for download characters.

There are some differences in the way they are implemented however. While download commands can be used to define a character between four and eleven columns of dots wide, dot graphics commands can be used to define a shape as narrow as one column of dots wide or as wide as 3264 dots on a Delta-15!

There is no "descender data" with dot graphics; graphics images are always printed with the top seven or eight pins of the print head, depending on whether you have a 7-bit or 8-bit interface (if you're not sure which type of interface your computer has, check the appendix for your computer).

So when do you use graphics and when do you use download characters? Practically anything you can do with graphics you can do with download characters, and vice versa. A clever programmer could actually plot a mathematical curve using download characters or use strings of graphics data as user-defined characters. But why do it the hard way? There are several instances when dot graphics is clearly the best way to approach the problem:

- If the graphic image to be printed is wider than 11 dots or higher than 7 dots
- If an image is to be printed just one time, as opposed to a frequently used "text" character
- If you want higher resolution (Delta can print as many as 240 dots per inch in dot graphics mode; text mode, which includes download characters, prints 60 dots per inch)

Using the Dot Graphics Commands

The command to print normal density (60 dots per inch horizontal; 72 dots per inch vertical) dot graphics uses this format:

`<ESC> "K" n1 n2 m1 m2. . .`

Just like many of the other codes you have learned, the command starts with an escape sequence (<ESC> "K" in this case). But unlike Delta's other codes there can be any number of graphics data bytes following the command. That's where n1 and n2 come in; they are used to tell Delta how many bytes of graphics data to expect.

Specifying the number of columns of dots

To figure the values of $n1$ and $n2$, you'll need to figure out how wide your graphic image will be (remember that there are 60 columns of dots per inch in normal density). Then comes the fun part: converting one number (the number of columns of dots) into two! Why is it necessary to use two numbers to tell Delta the number of graphics codes to expect? Because the largest number we can send in one byte (that's what the BASIC CHR\$() function sends: one byte) is 255. And with normal density graphics it's possible to have a graphics image as wide as 480 dots on Delta-10 or 816 dots on Delta-15. So to figure out how many columns of graphics data to expect, Delta multiplies $n2$ by 256 and adds the value of $n1$. If you divide the number of columns by 256, then $n2$ is the quotient and $n1$ is the remainder (why not let your computer figure it out for you: if the number of columns is assigned to variable X , then $N1 = X \text{ MOD } 256$ and $N2 = \text{INT}(X/256)$). Table 8-1 might make things even easier.

Table 8-1
Calculating $n1$ and $n2$

If the number of columns, x , ranges from:	then $n1$ is:	and $n2$ is:
1 to 255	x	0
256 to 511	$x - 256$	1
512 to 767	$x - 512$	2
768 to 1023	$x - 768$	3
1024 to 1279	$x - 1024$	4
1280 to 1535	$x - 1280$	5
1536 to 1791	$x - 1536$	6
1792 to 2047	$x - 1792$	7
2048 to 2303	$x - 2048$	8
2304 to 2559	$x - 2304$	9
2560 to 2815	$x - 2560$	10
2816 to 3071	$x - 2816$	11
3072 to 3264	$x - 3072$	12

Specifying the graphics data

Now that we've told Delta data how much data to expect, we better figure out how to send that information! Just as you do with download characters, with dot graphics you have control over the firing of every single pin on Delta's print head. In Figure 8-1, you can see that we've labeled each pin on the print head with a number, as we did with download characters (you should note one

important difference: this time the top pin has the highest value; for download character definitions it is the bottom pin). And specifying pins to fire is done in the same way: to fire the second pin from the top, for instance, send a CHR\$(64). Firing several pins at once is done in a similar fashion. For example, to print the first, third, and fourth dots, add their values (128 + 32 + 16) to send this total: CHR\$(176).




Figure 8-1. Starting with the most significant bit at the top, each pin of the print head is assigned a value which is a power of two. Note that for 7-bit computers, the top pin has a value of 64, and the bottom two pins are unused.

A short program should demonstrate how to implement the graphics command. The program below gave us this printout:

```

10 'PRINT GRAPHIC PATTERN
20 WIDTH "LPT1:",255
30 LPRINT CHR$(27) CHR$(75) CHR$(94) CHR$(1);
40 FOR I = 1 TO 25
50 FOR J = 0 TO 6
60 LPRINT CHR$(2^J);
70 NEXT J
80 FOR J = 6 TO 0 STEP -1
90 LPRINT CHR$(2^J);
100 NEXT J
110 NEXT I

```



```
120 WIDTH "LPT1:",80
130 LPRINT
```

In line 30 we've selected normal density graphics and said that 350 characters of graphics data would follow ($94 + (1 * 256) = 350$). The loop between lines 40 and 110 is repeated 25 times; this is what gives us the "zigzag" effect. The loop from line 50 to line 70 creates the lines that slope up; the loop between lines 80 and 100 prints the downward sloping lines. This is an example of plotting a very simple mathematical function to create a design. Later in this chapter we'll show something more complex.

Combining text and graphics

It's also possible to mix text and graphics in one line. This can be useful for labeling charts or graphs, or even inserting fancy graphics in text. Try adding these lines to our program:

```
25 LPRINT "WOW!";
115 LPRINT "THIS IS GREAT!";
```

Now if you run the program you should get a printout that looks like this:

```
WOW! ~~~~~THIS IS GREAT!
```

But there is one thing to be careful of: all graphics data must print on the same line. The graphics command is turned off at the end of each line, even if you have specified that more graphics codes follow. To see what we mean, change line 25 as shown and run the program.

```
25 LPRINT "WOW! THE DELTA-10 IS SIMPLY AMAZING.";
```

```
WOW! THE DELTA-10 IS SIMPLY AMAZING. ~~~~~
THIS IS GREAT!
```


(To get the same effect, Delta-15 users should change two program lines:

```
30 LPRINT CHR$(27) CHR$(75) CHR$(188) CHR$(2);  
40 FOR I = 1 TO 50
```

This will make the zigzag pattern long enough to go off the page.)

As you can see, Delta printed graphics up to the end of the line, then ignored the rest of the graphics data and returned to normal text on the next line.

Printing a Design or Logo

Since you control the firing of every pin, you can print nearly anything with Delta that you can draw (and probably better, if you're like most computer users!). This can be used for creating "computer art" or drawing maps. Or, as we'll show you here, you can use dot graphics to print your logo at the top of each letter you print.

Designing an image to print with dot graphics is much like designing download characters. The best way to start is to lay out your image on graph paper. Since you can print eight rows (seven with a 7-bit interface) of dots with each pass of the print head, draw a heavy horizontal line every eight rows on your graph paper. And it may be helpful to write the dot values (128, 64, 32, etc.) down the left side of each row. Then after you've filled in the "dots" that you want to print, it's time to get out the old calculator again! Just as you did with download characters, add up the values of each column of dots; this makes up one byte.

In the program below, we've taken the logo graphics information and put it into BASIC DATA statements. The program itself is short and simple. The loop starting at line 160 reads the data statements into a string array variable called LOGO\$. In line 230 we change the line spacing to 8/72 inch so that the lines of graphics data will connect vertically. The actual printing is done in the loop between lines 250 and 280; line 260 sends the graphics control code to Delta and line 270 sends one line of graphics data.

The printout from the program is shown right below the program.




Figure 8-2. By laying out the logo on graph paper, you can calculate all of the graphics data.

```

100 ' PRINT S&W LOGO
110 LINE.8$ = CHR$(27)+CHR$(65)+CHR$(8) 'Set line
    spacing to 8 dots
120 LINE.12$ = CHR$(27)+CHR$(50) 'Set line
    spacing to 1/6"
130 GRAPHIC$ = CHR$(27)+CHR$(75) 'Select dot
    graphics
140 DIM LOGO$(4)
150 WIDTH "LPT1:",255
160 ' READ DATA
170 FOR ROW = 1 TO 4
180 FOR COLUMN = 1 TO 100
190 READ P
200 LOGO$(ROW) = LOGO$(ROW) + CHR$(P)
210 NEXT COLUMN
220 NEXT ROW
230 ' PRINT LOGO
240 LPRINT LINE.8$;
250 FOR ROW = 1 TO 4
260 LPRINT GRAPHIC$;CHR$(100);CHR$(0);
270 LPRINT LOGO$(ROW)
280 NEXT ROW

```



```

290 LPRINT LINE.12$
300 'ROW 1
310 DATA 0,0,0,0,1,3,7,7,7,15,14,14,14,
      14,14,7,7,3,3,15
320 DATA 15,15,0,0,0,0,0,0,0,0,
      0,1,3,3,7,7,15,14,14,14
330 DATA 14,15,7,7,7,3,0,0,0,0,0,0,0,0,0,0
340 DATA 0,6,7,7,7,7,7,7,7,6,0,0,7,7,7,7
350 DATA 7,7,7,7,7,0,0,7,7,7,7,7,7,7,7,0
360 ' ROW 2
370 DATA 0,0,60,255,255,255,255,255,143,15,7,7,7,7,
      3,3,3,131,193,241
380 DATA 240,240,0,0,0,0,0,0,0,1,121,253,
      253,255,255,255,143,7,7,7
390 DATA 31,253,252,248,248,240,192,0,7,15,
      31,31,15,7,3,0,0,0,0,0
400 DATA 0,0,0,224,255,255,255,255,255,31,0,
      0,0,1,3,31,255,255,255,255
410 DATA 255,255,1,0,0,0,1,7,31,255,252,240,192,
      128,0,0,0,0,0,0
420 'ROW 3
430 DATA 0,31,31,3,129,128,192,192,192,192
440 DATA 192,224,224,224,224,240,255,255,255,255
450 DATA 255,127,0,0,0,0,63,127,255,255
460 DATA 255,255,193,128,128,128,128,192,224,240
470 DATA 252,255,255,255,127,63,31,7,7,31
480 DATA 254,252,248,224,128,0,0,3,7,7,7,3,0,0,
      192,255,255,255,255,255
490 DATA 15,15,63,252,240,192,0,240,255,255
500 DATA 255,255,255,7,15,127,252,240,192,0,0,
      0,0,0,0,0,0,0,0
510 'ROW 4
520 DATA 0,248,248,240,224,224,112,112,56,56
530 DATA 56,56,56,120,120,240,240,224,224,192,128,
      0,0,0,0,192,224,240,240
540 DATA 240,248,248,248,120,120,56,56,56,56
550 DATA 48,112,224,224,224,224,240,240,248,248
560 DATA 120,120,56,56,56,56,120,240,224,224
570 DATA 192,128,0,0,0,128,248,248,248,248,240,
      192,0,0,0,0,0,0,240,248
580 DATA 248,248,248,240,192,0,0,0,0,0,0,0,0,
      0,0,0,0,0,0,0

```


S&W

Plotting with Delta

This section of the manual gets into more serious BASIC programming just because it's required in order to have the computer act as a plotter driver. Don't be intimidated; while it's beyond the scope of this manual to teach BASIC, if you try the examples and take it slowly you should be doing some fancy plotting of your own before you know it.

If designing and calculating dot graphics images by laying them out on graph paper seems too tedious to you, then let the computer do the work for you! With your computer doing the calculations and Delta plotting the output, you can come up with some terrific business graphs, charts, and mathematical function plots.

The best way to do this is to set up an array in memory. This is your "graph paper." The first thing to do is to determine how big you want your output to be; this will determine the size of your array. (If you have grandiose plans to fill an entire page with plotter output, you better have lots of memory in your computer. With 60 dots per inch horizontally and 72 dots per inch vertically, it takes at least 540 bytes of memory for each square inch of plotted area. That doesn't sound so bad—but an area 8 inches square requires over 32K!)

Your array should be two-dimensional (just like graph paper) where one dimension will be the number of columns of dots and the other dimension is the number of printing lines (remember that you can have up to eight rows of dots per printed line).

Here's a program that will use calculated-shape graphics to plot a circle. As you'll see, by changing a few lines it can be used to plot virtually any shape.

```
1000 ' DELTA-PLOT
10000 ' Set program constants
1010 MAXCOL% = 75      : MAXROW% = 14
1020 DIM BIT%(MAXCOL%,MAXROW%)
1030 MASK%(1) = 64      : MASK%(4) = 8
1040 MASK%(2) = 32      : MASK%(5) = 4
1050 MASK%(3) = 16      : MASK%(6) = 2
1060 LX = 20           : LY = 20
```




```

1070 LXFAC = 72/LX      : LYFAC = 87/LY
2000 ' Plot curve
2010 RAD = 9
2020 X1 = 19   : Y1 = 10
2030 FOR ANG% = 0 TO 360 STEP 10
2040 RANG = ANG%*6.28/360
2050 X2 = RAD*COS(RANG)+10 : Y2 = RAD*SIN(RANG)+10
2060 GOSUB 4000
2070 NEXT ANG%
3000 ' Send bit image map to printer
3010 LPRINT CHR$(27) "A" CHR$(6)
3020 FOR ROW% = 0 TO MAXROW%
3030 A$ = ""
3040 LPRINT CHR$(27) "K" CHR$(MAXCOL%) CHR$(0);
3050 FOR COL% = 1 TO MAXCOL%
3060 A$ = A$ + CHR$(BIT%(COL%,ROW%))
3070 NEXT COL%
3080 LPRINT A$ " "
3090 NEXT ROW%
3100 LPRINT CHR$(27) "2"
3110 END
4000 ' Draw a line from X1,Y1 to X2,Y2
4010 XL = X2 - X1      : YL = Y2 - Y1
4020 NX = ABS(XL*LXFAC) : NY = ABS(YL*LYFAC)
4030 IF NX < NY THEN NX = NY
4040 NS% = INT(NX+1)
4050 DX = XL/NS%       : DY = YL/NS%
4060 FOR I% = 1 TO NS%
4070 X1 = X1 + DX      : Y1 = Y1 + DY
4080 GOSUB 5000
4090 NEXT I%
4100 RETURN
5000 ' Plot a point at X1,Y1
5010 XX = X1 * LXFAC   : YY = Y1 * LYFAC
5020 COL% = INT(XX) + 1
5030 ROW% = INT(YY/6)
5040 XIT% = INT(YY - ROW% * 6)+1
5050 BIT%(COL%,ROW%) = BIT%(COL%,ROW%) OR
    MASK%(XIT%)
5060 RETURN

```

How the program works

In the program above, we've created an array called BIT%, which is dimensioned in line 1020. You'll note that instead of



using numeric constants to dimension the array, we used the variables MAXCOL% and MAXROW%. This way, if your computer has enough memory and you want to plot a larger image, all you need to change are the values in line 1010. The array MASK% contains the values of the dots. (In order to make this program run on the most computers, we're using only six pins for graphics. With many computers, you can use all eight available pins.) In lines 1060 and 1070 we've defined some other variables you'll be interested in: LX, LXFAC, LY, and LYFAC are used as scaling factors. By changing these values, you can change the size of your printed image or even distort it (you can, for example, make our circle print as an ellipse). Experiment a little bit!

The main calculations for plotting the image are done starting at program line 2000. This is where you put the formulas that you want to plot. By changing just the lines between 2000 and 3000 (with some creative mathematics!) you can plot any function—limited only by your imagination. Some examples are shown at the end of this section.

What the program section starting at 2000 actually does is to calculate starting and ending points for a line (in our circle the "lines" are very short—sometimes the starting and ending points are the same). The coordinates of the starting point of the line are assigned to variables X1 and Y1. The line ends at point X2,Y2. When these coordinates have been calculated, a subroutine call is made to line 4000. This subroutine calculates the coordinates of individual points along that line.

After these coordinates have been determined, the subroutine at line 5000 is called. This routine turns "on" an individual dot in our array called BIT%. (Keep in mind that no printing has been done yet; the computer is still drawing the image on its "graph paper" in memory.) The way an individual dot is turned on is using the logical OR function in line 5050.

When all the points have been plotted in memory, printing begins at line 3000. We first set the line spacing to 6/72 inch using the <ESC> "A" command. This is so that there are no gaps between rows of dots. Then the loop from line 3020 to line 3090 prints the dot graphics image one line (which is six dots high) at a


time. The variable A\$ is used to build a string of all the columns of BIT% in a given row.

As you can see, by taking the program in small pieces and analyzing it, graphics programming does not have to be difficult. If you want to try some other plots, try these (replace lines between 2000 and 3000 with the lines below). The printouts from each program are shown below the listing.

```

2000 ' Plot curve
2010 RAD = 9
2020 FOR ANG% = 0 TO 360 STEP 15
2030 RANG = ANG%*6.28/360
2040 RANG2 = (ANG%+150)*6.28/360
2050 X1 = RAD*COS(RANG)+10 : Y1 = RAD*SIN(RANG)+10
2060 X2 = RAD*COS(RANG2)+10 : Y2 =
      RAD*SIN(RANG2)+10
2070 GOSUB 4000
2080 NEXT ANG%


```



```

2000 ' Plot curve
2010 X1 = 0 : Y1 = 10 : X2 = 20 : Y2 = 10 : GOSUB
      4000
2020 X1 = 10 : Y1 = 0 : X2 = 10 : Y2 = 20 : GOSUB
      4000
2030 X1 = 0 : Y1 = 10 : FOR X2 = 0 TO 20 STEP .2
2040 Y2 = 10 - 9 * SIN(3.14159 * X2/10) : GOSUB
      4000
2050 NEXT X2

```

Using Delta for business graphics

You don't have to be a mathematician, scientist, or computer hacker/artist to use Delta's graphics capabilities. It can be used for business graphics too—line graphs, bar charts, pie charts, and more! There are many commercially available graphics programs that support Delta's graphics. And, of course, you can write your own. To get you started, we've written a program that prints a pie chart. Here it is:

```

100 ' PIECHART
110 ESC$ = CHR$(27)      : LF$ = CHR$(10)
120 FF$ = CHR$(12)      : VTAB$ = CHR$(11)
130 EMPHASIZED$ = ESC$ + "E" : NOT.EMPHASIZED$ =
    ESC$ + "F"
1000 ' Set program constants
1010 OPEN "LPT1:" AS #1 : WIDTH #1,255
1020 DIM BIT%(190,36),A$(36),PCT%(25),TEXT$(42),
    PIECETEXT$(25)
1030 MASK%(1) = 64      : MASK%(4) = 8
1040 MASK%(2) = 32      : MASK%(5) = 4
1050 MASK%(3) = 16      : MASK%(6) = 2
1060 LX = 20            : LY = 20
1070 LXFAC = 190/LX     : LYFAC = 216/LY
1080 FOR I= 1 TO 42
1090 TEXT$(I) = SPACE$(80)
1100 NEXT I
1110 GOSUB 7000
2000 ' Plot curve
2010 RAD = 9
2020 X1 = 19            : Y1 = 10
2030 PRINT " ";
2040 FOR ANG% = 0 TO 360 STEP 5
2050 RANG = ANG%*6.28/360
2060 X2 = RAD*COS(RANG)+10 : Y2 = RAD*SIN(RANG)+10
2070 GOSUB 4000

```



```

2080 NEXT ANG%
2090 FOR PIECE% = 1 TO NUMBER.PIECES%
2100 X1 = 10 : Y1 = 10
2110 TOTAL.PCT%=TOTAL.PCT%+PCT%(PIECE%)
2120 ANG%=360*TOTAL.PCT%*.01
2130 RANG = ANG%*6.28/360
2140 X2 = RAD*COS(RANG)+10 : Y2 = RAD*SIN(RANG)+10
2150 GOSUB 4000
2160 GOSUB 6000
2170 NEXT PIECE%
3000 ' Send bit image map to printer
3010 PRINT
3020 FOR ROW% = 0 TO 35
3030 A$(ROW%) = ""
3040 FOR COL% = 1 TO 190
3050 A$(ROW%) = A$(ROW%) + CHR$(BIT%(COL%,ROW%))
3060 NEXT COL%
3070 PRINT CHR$(176);CHR$(176);
3080 NEXT ROW%
3090 PRINT
3091 PRINT#1,SPACE$(40-LEN(TITLE$)/2);EMPHASIZED$;
      TITLE$;NOT.EMPHASIZED$;LF$
3100 PRINT#1,VTAB$;VTAB$;VTAB$
3110 PRINT#1,ESC$;"A";CHR$(6)
3120 PRINT#1,TEXT$(1);LF$;TEXT$(2);LF$;TEXT$(3);LF$
3130 FOR ROW% = 0 TO 35
3140 PRINT#1," " ;ESC$;"K";
      CHR$(190);CHR$(0);
3150 PRINT#1,A$(ROW%)
3160 PRINT#1,TEXT$(ROW%+4);LF$
3170 PRINT CHR$(176);CHR$(176);
3180 NEXT ROW%
3190 PRINT#1,TEXT$(40);LF$
3200 PRINT#1,TEXT$(41);LF$
3210 PRINT#1,TEXT$(42);LF$
3220 PRINT#1,ESC$;"2";FF$
3230 END
4000 ' Draw a line from X1,Y1 to X2,Y2
4010 XL = X2 - X1 : YL = Y2 - Y1
4020 NX = ABS(XL*LXFAC) : NY = ABS(YL*LYFAC)
4030 IF NX < NY THEN NX = NY
4040 NS% = INT(NX+1)
4050 DX = XL/NS% : DY = YL/NS%
4060 FOR I% = 1 TO NS%
4070 X1 = X1 + DX : Y1 = Y1 + DY

```



```
4080 GOSUB 5000
4090 NEXT I%
4100 PRINT CHR$(29);CHR$(205);CHR$(175);
4110 RETURN
5000 ' Plot a point at X1,Y1
5010 XX = X1 * LXFAC      : YY = Y1 * LYFAC
5020 COL% = INT(XX) + 1
5030 ROW% = INT(YY/6)
5040 XIT% = INT(YY - ROW% * 6)+1
5050 BIT%(COL%,ROW%) = BIT%(COL%,ROW%) OR
      MASK%(XIT%)
5060 RETURN
6000 ' Place text fields in the proper location
6010 MIDANG%=(ANG%+PREVANG%)/2
6020 RANG = MIDANG%*6.28/360
6030 X3 = INT(20*SIN(RANG)) : Y3 =
      INT(22*COS(RANG))
6040 X4 = 22 + X3          : Y4 = 40 + Y3
6050 IF MIDANG%>270 OR MIDANG%<90 THEN
      MID$(TEXT$(X4),Y4) = PIECETEXT$(PIECE%) ELSE
      MID$(TEXT$(X4),Y4-LEN(PIECETEXT$(PIECE%)))=
      PIECETEXT$(PIECE%)
6060 PREVANG%=ANG%
6070 RETURN
7000 ' Accept Data from Screen
7010 CLS: PRINT : PRINT : PRINT :
7020 INPUT "ENTER TITLE FOR CHART: ",TITLE$
7030 AMT.SOFAR%=0          : AMT.LEFT%=100
7040 FOR I=1 TO 24
7050 CLS: PRINT "          ENTER PARAMETERS FOR
      PIE-CHART"
7060 PRINT " TOTAL SO FAR : ";
7070 PRINT USING "###";AMT.SOFAR%
7080 PRINT " TOTAL REMAINING: ";
7090 PRINT USING "###";AMT.LEFT%
7100 PRINT :PRINT :PRINT :PRINT
7110 INPUT "ENTER PERCENTAGE FOR FIELD:  ",PCT%(I)
7120 IF PCT%(I)>AMT.LEFT% OR PCT%(I)=0 THEN
      PCT%(I)=AMT.LEFT%
7130 AMT.LEFT%=AMT.LEFT%-PCT%(I)
7140 AMT.SOFAR%=AMT.SOFAR%+PCT%(I)
7150 PRINT :PRINT
7160 INPUT "ENTER DESCRIPTION OF FIELD:  ",
      PIECETEXT$(I)
```




```
7170 IF LEN(PIECETEXT$(I))>15 THEN PRINT "FIELD TOO  
LONG - 15 CHAR. MAX": GOTO 7160  
7180 IF AMT.LEFT%=0 GOTO 7200  
7190 NEXT I  
7200 NUMBER.PIECES%=I  
7210 IF NUMBER.PIECES%=1 THEN 7030  
7220 CLS  
7230 RETURN
```

You should recognize many sections of code from the plotting program. We've just expanded on that program framework to include routines for inputting data to be graphed (starts at line 7000) and placing labels next to the pie chart (starts at line 6000).

The output from our program is shown below.

Sales by region



High Resolution Graphics

Up until now all of the dot graphics printing we have done has been with Delta's normal density mode. This can give you some pretty sharp images at great speed. Sometimes though, you may want to create an image with even higher resolution. Delta has four density modes you can use; they're summarized in Table 8-2.


Table 8-2
Dot graphics commands

Function	Control code
Normal density (60 dots/inch)	⟨ESC⟩ "K" n1 n2 m1 m2 . . .
Double density (120 dots/inch)	⟨ESC⟩ "L" n1 n2 m1 m2 . . .
Double density/double speed	⟨ESC⟩ "y" n1 n2 m1 m2 . . .
Quadruple density (240 dots/inch)	⟨ESC⟩ "z" n1 n2 m1 m2 . . .


Note: If your computer does not support lowercase characters, use CHR\$(121) and CHR\$(122) for "y" and "z", respectively.

The command syntax for all of the commands is the same—just as you have learned it for the ⟨ESC⟩ "K" (normal density) command. The number of columns to be printed is $n1 + 256 * n2$.


So what do these different modes do? On the following pages are actual size reproductions of printouts of the same image printed in each of the four different densities. They were all printed using the plotting program in this chapter (with a rather complex set of formulas starting at line 2000!).




Normal density graphics



Double density graphics



Double density/double speed



Quadruple density graphics

So if quadruple density looks so great, why not use it all the time? Let's try an experiment on your printer which will show just how the different density modes work. Using the first program in this chapter, change line 30 to try each of the different modes. Just change the CHR\$(75) (which is the ASCII code for "K") to "L", "y", and "z" in turn. Your printouts should look something like this:

~~~~~ (ESC)"L"

~~~~~ (ESC)"y"

~~~~~ (ESC)"z"

As you can see, the different modes seem to condense the printed image. So, to get the same image in a higher density mode, you must plot more points. This requires twice as much memory for your array, twice as much computing time, and twice as much printing time (but the results may be worth it!).

Star's engineers have given programmers a unique shortcut for program development though—double density double speed mode. Although this mode requires just as much memory and computing time as double density, it prints at the same speed as normal density graphics. Amazing, you say? Well, it is—until you know the secret. Every other column of dots is ignored, so the output is actually the same as normal density graphics. The advantage is that you can write and debug your programs at double speed, then change to double density graphics for terrific output.

## ***If You Have Problems with BASIC***

You may write some graphics programs that look just right in the listing, but the printouts aren't quite what you expected. A common problem is that the BASIC interpreter in your computer is inserting a few of its own codes. For instance, if your program generates a CHR\$(13) as valid graphics data, BASIC may follow it




with a CHR\$(10). Another problem arises with certain computers that replace horizontal tabs (CHR\$(9)) with a series of spaces (CHR\$(32)). A possible solution to these problems is to not use the bottom dot (which has a value of 1). This way, you will never produce an odd number, hence, you will never have a CHR\$(13) or CHR\$(9). (This is why we used only six pins in our plotting program.)

That's one solution to one problem. You'll find more of each (with specific information for your computer) in the appropriate appendix.

## Summary

| Control code                | Function                                             |
|-----------------------------|------------------------------------------------------|
| <ESC> "K" n1 n2 m1 m2 . . . | Print n1 + 256*n2 columns of normal density graphics |
| <ESC> "L" n1 n2 m1 m2 . . . | Print double density graphics                        |
| <ESC> "y" n1 n2 m1 m2 . . . | Print double density graphics at double speed        |
| <ESC> "z" n1 n2 m1 m2 . . . | Print quadruple density graphics                     |





## **Chapter 9**

# ***Getting it All Together***

You have now seen how all of Delta's functions work. You surely have some good ideas of ways to use some of Delta's many capabilities. With all the different printing styles available you won't run out of variations for quite a while. And if there is a style of printing that Delta doesn't have built-in, you can develop your own with Delta's download characters.

Four different modes of graphics provide you with limitless ways to create pictures with Delta. You have learned how to create both stored-data and calculated shapes using Delta's graphics capabilities.

The calendar on the following page is a demonstration of just some of Delta's abilities. You will recognize many of the things




that you have learned while reading this book. The globe was created the same way as the calculated-shape graphics that you learned about. The S&W logo was used to illustrate stored-data graphics. And you are sure to recognize many of the print styles used at the top of the calendar. Those shaded bars are created by using different graphics densities.

The numbers in the calendar itself are the ones that you created with download characters. And the lines creating the boxes are made with Delta's block graphics characters.

Delta's flexibility in line spacing and its ability to mix many types of printing on one line make it possible to create complex forms like this calendar. With Delta's advanced features it's easy to create a business form or letterhead, and fill it out at the same time. That's a productive printer.



**S&W****Smith & Williams**

123 Burritt Street

Hackensack, New Jersey 07602


*Microcomputer Consultants to the World***JANUARY 1984**

| SUNDAY | MONDAY | TUESDAY | WEDNESDAY | THURSDAY | FRIDAY | SATURDAY |
|--------|--------|---------|-----------|----------|--------|----------|
| 1      | 2      | 3       | 4         | 5        | 6      | 7        |
| 8      | 9      | 10      | 11        | 12       | 13     | 14       |
| 15     | 16     | 17      | 18        | 19       | 20     | 21       |
| 22     | 23     | 24      | 25        | 26       | 27     | 28       |
| 29     | 30     | 31      |           |          |        |          |









## **Chapter 10**

# **Maintenance**

As almost any good mechanic will tell you, dust and heat are the prime enemies of any mechanism, and Delta is no exception. The best maintenance is preventive. So, to start with, we hope you've found a clean, dust-free location with a comfortable temperature range for your computer printer system, and that you'll keep the printer's dust-cover where it belongs — in its place atop the printer! Appendix A gives you further tips on locating Delta.

## **Cleaning Delta**

The second rule for long life is *periodic cleaning*. Both inside



and outside of the case respond gratefully to periodic cleaning with a damp rag and alcohol. Do this whenever the case appears to be getting dirty, always being careful to avoid dripping alcohol on the printer mechanism. To clean the inside of dust and paper lint, use a nice soft brush, after first removing the printer's upper case in the following manner.


## Removing the Upper Case

First, turn the power switch off and unplug Delta.

Remove the tractor unit.

Remove the platen knob.

Next, remove the two Phillips screws, per Figure 10-1. (You'll find three screws on the larger Delta-15.)




**Figure 10-1.** To remove Delta's upper case, remove the screws.

Lift up the upper case from the back; then "pop out" the front of the case by pushing outward with your thumbs where the plastic tabs fit into the bottom case.

Then stand the upper case on its end, right next to the printer so the short colored wires attached to the control panel remain connected. It's a good idea to keep the case from falling over by



setting the corner of the printer on the edge of the case as shown in Figure 10-2.



**Figure 10-2.** Tilt the cover up so that you don't have to disconnect the control panel wires.


**NOTE TO THE UNWARY:** When cleaning the inside of Delta, be very careful not to bend or injure any of those tiny wires or other little gizmos that lie exposed and defenseless against a heavy-handed touch.

Besides the periodic cleanings, the only other maintenance you'll encounter will be changing the ink ribbon (see Appendix A for these directions), and replacing an occasional blown fuse or replacement of the print head after a long period of use.

## ***Replacing a Fuse***

How can you tell when you've blown a fuse? Well, when the printer won't operate and the power lamp on the control panel isn't lit, even though you're sure the power switch is on and the printer is plugged in — it's likely a blown fuse.





**Figure 10-3.** The secondary fuses are a little tricky to reach.

To check out the fuses (there are two), turn the power switch off, and disconnect the power plug. Then remove the platen knob and the upper case, as described in the preceding section.

When the case is off, check for the correct fuse locations on Figure 10-3. Remove the primary fuse first, held in its clamps near the power switch. The fuse is a commonly used type, with a metal strip suspended in a glass and metal case. If the strip is broken, the fuse is blown.



Replace this fuse with a 2A/125V slow-blow type fuse; then test-run the printer. Still immobile? Check the secondary fuse, which is still further inside the “works,” as shown in Figure 10-3. If that’s blown, replace it with a 3A/125V slow-blow type fuse.

Then if the printer still isn’t working, better call on your Delta dealer for help.

Incidentally, it makes it easier to remove and replace the fuses if you have a long, flat screwdriver and some needlenose pliers handy. The fuses are small, and the secondary fuse is sort of hard to get at without these particular tools. (But be careful not to crush the glass fuses with the pliers.)

We’ve found the following technique works pretty well for replacing the secondary fuse. Using the long flat screwdriver, pop one end of the fuse loose from its holding clamp and stand it on end. Then grab the loose end with the needle nose pliers and pull it out. To insert the new fuse, lower it into place with the pliers, so that it lies flat against the two holding clamps. Then use the long screwdriver to push each metal end of the fuse into its adjacent holding clamp.


## ***Replacing the Print Head***

As mentioned earlier, the print head has a long life, printing perhaps 100,000,000 characters before it wears out. You’ll know when that happens when the printout is too faint for your taste even after replacing the ink ribbon.

**WARNING:** The print head gets hot during operation, so let it cool off for awhile, if necessary, to avoid burning your fingers.

To replace the print head, start by removing the printer cover and the ink ribbon. Then unplug the print head cable (see Figure 10-4) while holding down the head cable board. Next, remove the two screws and washers which fasten the print head. Then place the new print head in position, and attach it with the same two screws. Apply “screw lock,” (an adhesive available at hardware stores) to the heads of the screws. Finally, insert the head cable, with the printed side up, tightly into the head connector on the cable board. Be sure that it’s a good solid connection, or it could cause problems.






**Figure 10-4.** Replacement of Delta's print head is simple.

Back to perfect printing!





## ***Appendix***







## **Appendix A**

# **Setting Up Delta**

In this appendix, we'll show you how to unpack your new Delta printer, set it up in the right location, and get it ready for you to load it with paper and start printing! But first . . .

### **Where shall we put it?**

Before you do anything else, give some thought to where you'll be using your printer. Obviously, it will be somewhere near your computer. And both computer and printer will lead longer, healthier lives if they like their environment. For instance, we recommend . . .

- Placing the printer on a flat surface
- Keeping it out of direct sunlight and away from heat-producing appliances
- Using it only in temperatures where you are comfortable
- Avoiding areas with a lot of dust, grease, or humidity
- Giving it "clean" electricity. Don't connect it to the same circuit as large, noise-producing motors
- Providing the right voltage, which is not over 10% more or less than 120 volts AC.

**WARNING:** Extremely high or low voltage can damage your printer.

### **What have we here?**


Let's take a look at what's in the box. Take it slow and easy, and check each item in the box against Figure A-1. There should be exactly 11 items. One important item is the printer registration and warranty card. Now is the time to fill it in and mail it. It's a good warranty, and you'll like the protection it gives you.

Let's move on to the next step . . .

### **Removing the shipping screws**

There are two shipping screws on the bottom of the printer,






**Figure A-1.** Inside the carton you should find: 1) printer, 2) printer cover, 3) paper guide, 4) paper separator, 5) roll paper holder, 6) roll paper shaft, 7) ink ribbon, 8) platen knob, 9) spare fuses, 10) User's Manual, 11) warranty card.

used to hold the internal chassis to the external frame during shipping. To get at these, carefully place the printer upside down on a soft surface like a foam cushion. Remove the two screws with a Phillips screwdriver as shown in Figure A-2.





**Figure A-2.** Remove the shipping screws before using Delta.

### ***Removing the packing from inside the printer***

Remove three cardboard pieces: a large flat piece protecting the print head, and two smaller pieces stuck in at either end of the platen (the rubber cylinder that feeds the paper through the printer).

You'll want to save the screws, along with the rest of the packing material and the shipping box, in case you ever have to ship the printer. Tape the screws somewhere on the box or packing. (You *did* fill in that registration card, didn't you?)


### ***Installing the platen knob***

This is the knob that turns the rubber platen cylinder. It fits into the hole on the right side of the printer case. Just match the odd-shaped hole in the knob with the same shape on the shaft you'll see inside the hole in the case, and press on firmly. Give the knob a few twirls to see that it's turning the platen easily and smoothly.

### ***Removing the tractor unit***

The tractor unit, shown in Figure A-4, comes mounted on the printer during shipment. It is used only with sprocket-feed paper. When other papers are used, such as single sheets or roll paper,





**Figure A-3.** There are three pieces of cardboard to remove.

the unit should be removed, in the following manner:

Remove the printer cover (if attached).

Identify the "snap levers" as shown in Figure A-4.

Pull both snap levers forward, and at the same time . . .

Rock the tractor unit up and towards you about half an inch.


Now lift the tractor up and away from the printer.

Up to this point, we've been clearing the decks for action, so to speak. Only two more things are left to do before we can start printing. They are, 1) attach the paper separator and paper guide racks, and 2) install the ink ribbon. Actually, if you're planning to print on single sheets only, you won't need to use the paper separator and paper guide, which are designed expressly to guide continuous paper (roll or sprocket-feed) through the printer.


### ***Attaching the paper separator and paper guide***

First, identify the paper separator (the large metal rack), shown in Figure A-5. Insert one protruding end into the hole shown in the diagram, then gently bend the other protruding end until it snaps into the opposite hole. Follow the same procedure with the smaller metal rack, which is the paper guide.





**Figure A-4.** Remove the tractor by pulling the snap levers towards you and tilting the tractor unit back.



**Figure A-5.** Attach the paper guide and paper separator.




Important news: If you get these in upside down, they won't work. So take another sharp look at Figure A-5 before we pass on to the final act—installing the ink ribbon.

(NOTE: If you're wondering about the wire roll paper holder and holder shaft, we'll explain these in Chapter 1, where we discuss the whole subject of paper selection, paper feed, and related topics.)

### **Installing the ribbon**

Telling you how to set the ribbon is like writing a set of instructions on how to tie your shoelaces. It takes a lot longer to describe it than it does to do it! So, you'll be smart to study the several figures shown here; they'll tell you all you really need to know.


Nevertheless, if you feel better following written instructions, read on . . .




**Figure A-6.** You'll find this diagram of the ink ribbon path inside your Delta for easy reference when you change ribbons.

After looking carefully at the Figure A-6, begin by turning the power off, and removing the printer cover. Then slide the print





**Figure A-7.** The eyelet should be between the ribbon spool and the changeover lever.



**Figure A-8.** You may want to use a ballpoint pen to press the ribbon guide towards the platen and work the ribbon down beside it.



head gently to the center of the printer. Next, set one of the ribbon spools on a ribbon spool post, being careful to have it wind/unwind exactly as shown in the figures. The spool should fit so that the drive pins engage the spool.

Next, you start threading the ribbon. The only tricky part is around the print head, so pay particular attention to Figure A-8. The ribbon should fit in the slot between the print head and the ribbon guide. Use a ballpoint pen to work the ribbon down between the print head and the ribbon guide.

The other thing to watch is the position of the eyelet near the end of the ribbon. This works exactly like a typewriter ribbon eyelet; it can't get by the slotted guide, which causes the ribbon to automatically reverse its direction.

Finally, set the other spool snugly on the opposite spool holder; then turn the spools by hand four or five turns in each direction to verify that everything is properly set and ready to roll.

### ***Installing the printer cover***

Now that you've completed the steps described in this Appendix A, you may want to mount the printer cover in place to keep dust and dirt away. It's a simple procedure. Merely fit the two tabs at the back edge of the cover into the two slots right next to the two holes where you fastened the wire rack paper separator. Then drop the other end down gently until it sits firmly on the printer. That's all!

To remove the cover just reverse the process: lift up the front and pull it out of the slots at the back.

### ***Connecting Delta to your computer***

To complete the installation, you'll need to connect Delta to your computer. In Appendixes B through G, we've described the procedure, including specific guidelines for making connections with several of the most popular computers used by Delta owners.



## **Appendix B**

# **IBM Personal Computer and Compaq Computer**

Both the IBM Personal Computer and the Compaq computer function the same when connected to Delta. We will discuss the IBM-PC, knowing that all we say works just as well for the Compaq.

Delta can connect to either a serial or a parallel interface in the IBM-PC or IBM-XT computers. IBM calls a parallel interface a "Parallel Printer Adapter", and they call a serial interface an "Asynchronous Communications Adapter."

You only need a cable to connect Delta to your IBM-PC. Your Delta dealer can furnish this cable, or you can use a standard IBM-PC parallel printer cable for the parallel interface.

We recommend that you set the DIP switches in Delta as shown below when connecting it to an IBM-PC parallel interface.

**Recommended DIP Switch Settings for IBM-PC**

| <b>Switch</b> | <b>Setting</b> | <b>Function</b>           |
|---------------|----------------|---------------------------|
| 1-1           | ON             | 11 inch page size         |
| 1-2           | ON             | Normal print density      |
| 1-3           | ON             | 10 CPI pitch              |
| 1-4           | ON             | Normal characters         |
| 1-5           | ON             | 1/6 inch line feed        |
| 1-6           | ON             | U.S.A. Character set      |
| 1-7           | ON             |                           |
| 1-8           | ON             |                           |
| 2-1           | ON             | Paper-out detector active |
| 2-2           | OFF            | Parallel interface        |
| 2-3           | OFF            | 8-bit interface           |
| 2-4           | OFF            | No auto line feed         |



**IBM-PC Parallel Cable**

| Delta   |           | IBM-PC Parallel |           |
|---------|-----------|-----------------|-----------|
| Pin No. | Function  | Pin No.         | Function  |
| 1       | STROBE    | 1               | STROBE    |
| 2       | D1        | 2               | D0        |
| 3       | D2        | 3               | D1        |
| 4       | D3        | 4               | D2        |
| 5       | D4        | 5               | D3        |
| 6       | D5        | 6               | D4        |
| 7       | D6        | 7               | D5        |
| 8       | D7        | 8               | D6        |
| 9       | D8        | 9               | D7        |
| 10      | ACK       | 10              | ACK       |
| 11      | BUSY      | 11              | BUSY      |
| 12      | PAPER END | 12              | PAPER END |
| 13      | SELECTED  | 13              | SELECT    |
| 16      | GROUND    | 18-25           | GROUND    |
| 31      | RESET     | 16              | RESET     |
| 32      | ERROR     | 15              | ERROR     |

The IBM-PC expects its printer to be connected to the parallel interface. If you are using the serial interface, then you will need to instruct your computer to send information to the serial interface instead of to the parallel interface. This is done with the MODE command. You must use the following two commands each time you turn on your computer.

```
MODE COM1:48,N,8,1,P
MODE LPT1:=COM1:
```

The first "sets up" the asynchronous adapter to match the settings of DIP switch 3 in Delta. The second re-directs printer output to the serial port. The switches on DIP switch 3 must be set as shown below to use this MODE command. (The IBM-DOS manual tells you how to create a different MODE command for different DIP switch settings.) You can put these two MODE commands into a file named AUTOEXEC.BAT and it will execute automatically each time you start your computer.



**Table B-1**  
**Serial switch settings**

| Switch | Setting | Function                   |
|--------|---------|----------------------------|
| 3-1    | OFF     | 8 data bits                |
| 3-2    | OFF     | No parity                  |
| 3-3    | ON      | Serial busy,<br>block mode |
| 3-4    | OFF     |                            |
| 3-5    | either  | Parity                     |
| 3-6    | ON      | 4800 baud                  |
| 3-7    | ON      |                            |
| 3-8    | OFF     |                            |

The serial cable shown below will work with DIP switch 3 set as shown above to connect Delta to a serial interface on the IBM.

**IBM-PC Serial Cable**

| Delta   |                     | IBM-PC  |                 |
|---------|---------------------|---------|-----------------|
| Pin No. | Function            | Pin No. | Function        |
| 2       | TRANSMIT DATA       | 3       | RECEIVE DATA    |
| 3       | RECEIVE DATA        | 2       | TRANSMIT DATA   |
| 4       | REQUEST TO SEND     | 5       | CLEAR TO SEND   |
| 5       | CLEAR TO SEND       | 4       | REQUEST TO SEND |
| 7       | SIGNAL GROUND       | 7       | SIGNAL GROUND   |
| 8       | CARRIER DETECT      | 4       | REQUEST TO SEND |
| 20      | DATA TERMINAL READY | 6       | DATA SET READY  |

## **BASIC programing**

All the programs in this book are written in the BASIC used by the IBM-PC. That makes it easy to do the things that we show you. But when you start writing your own programs there are several things that you should know.

IBM BASIC defaults to a printer width of 80. This means that it will automatically insert a carriage return and line feed after every 80 characters. If you want to print lines longer than 80 characters you will need to change the width of the printer. If you set the printer width to 255, then the IBM will never insert a line feed and carriage return, unless you start a new line. (This is what you want usually.) To set the width of the printer to 255, use this statement:

```
100 WIDTH "LPT1:", 255
```



IBM BASIC has one other little trick that will mess up your graphics if you let it. IBM BASIC is very insistent about adding a line feed to a carriage return. This is fine if you are printing text, but if an ASCII 13 pops up in the middle of your graphics printout, IBM BASIC will *still* add a line feed to it. This will put strange things in the middle of your graphics, and leave you with extra characters at the end of your line.

There is an easy way to avoid this problem. You just open the printer as a random file. The following program shows how this is done.

```
10 OPEN "LPT1:" AS #1          ' RANDOM ACCESS
20 WIDTH #1, 255               ' SET WIDTH TO 255
30 PRINT #1, "TESTING"        ' PRINT A LINE
40 PRINT #1, CHR$(10)         ' ADD YOUR OWN LF
```

### ***Listing programs***

To list programs on the IBM-PC, use the LLIST command. This directs the listing to the printer instead of the screen.

### ***Program listings***

There are no program listings given here for the IBM-PC because all the programs in the book are written for the IBM-PC.



## Appendix C

# Apple II Computers

Apple II computers require an interface board (mounted inside the Apple II) and a cable to run Delta. Star recommends that you use the **grafstar™** interface for the Apple II, II+, and IIe, by Star Micronics. It comes complete with a cable and is easily installed. A unique feature of the **grafstar™** makes it possible to do some fancy dot graphics programming.

You can, of course, use many of the available parallel interface boards for the Apple II, and an appropriate cable.

### Setting the switches

We recommend that you set the DIP switches in Delta as shown below when connecting it to an Apple II.

**Recommended DIP Switch Settings for Apple**

| Switch | Setting | Function                  |
|--------|---------|---------------------------|
| 1-1    | ON      | 11 inch page size         |
| 1-2    | ON      | Normal print density      |
| 1-3    | ON      | 10 CPI pitch              |
| 1-4    | ON      | Normal characters         |
| 1-5    | ON      | 1/6 inch line feed        |
| 1-6    | ON      | U.S.A. Character set      |
| 1-7    | ON      |                           |
| 1-8    | ON      |                           |
| 2-1    | ON      | Paper-out detector active |
| 2-2    | OFF     | Parallel interface        |
| 2-3    | ON      | 7-bit interface           |
| 2-4    | OFF     | No auto line feed         |

### Applesoft BASIC

The Apple II computer, using Applesoft BASIC, does not



**Apple Parallel Cable**

| Delta   |          | Apple Board |          |
|---------|----------|-------------|----------|
| Pin No. | Function | Pin No.     | Function |
| 25      | SIG GND  | 1           | SIG GND  |
| 26      | SIG GND  | 2           | SIG GND  |
| 27      | SIG GND  | 3           | SIG GND  |
| 1       | STROBE   | 4           | STROBE   |
| 28      | SIG GND  | 5           | N/C      |
| 2       | DATA1    | 6           | DATA1    |
| 3       | DATA2    | 7           | DATA2    |
| 4       | DATA3    | 8           | DATA3    |
| 5       | DATA4    | 9           | DATA4    |
| 6       | DATA5    | 10          | DATA5    |
| 7       | DATA6    | 11          | DATA6    |
| 8       | DATA7    | 12          | DATA7    |
| 9       | DATA8    | 13          | DATA8    |
| 10      | ACK      | 14          | ACK      |
| 29      | SIG GND  | 15          | SIG GND  |

have different types of PRINT statements for the screen and printer. You must add commands to your programs that direct the output of the PRINT statements to the printer. To direct output to the printer (with the interface board in slot #1) you must use the PR# 1 command. Depending on the version of Applesoft BASIC that you are using this command can take various forms. It is usually one of the following:

```
10 PR# 1
or
10 PRINT "<Ctrl-D>PR#1"
or
10 PRINT CHR$(4) "PR#1"
```

To return output to the screen, the command is PR#0, in the same form that works for PR# 1.

To allow line lengths longer than the Apple II usually uses you must add the following statement to your programs:

```
20 PRINT CHR$(9) "255N"
```



This allows lines of any length to be sent to the printer and is especially important for dot graphics. (The number 255 in the BASIC statement above could be replaced by any number from 0 to 255 and would set the line length to that value.)

Two codes are a particular problem on the Apple II: CHR\$(7) and CHR\$(9). Try to avoid using these in dot graphics programs.

The Apple II computer uses CHR\$(9) as a printer initialization code. It won't send it on to the printer. There is a way to bypass this problem, however. You can change the printer initialization code to a value other than CHR\$(9) like this:

```
PR#1
PRINT CHR$(9); CHR$(1)
```

This makes CHR\$(1) the printer initialization code (and transfers the problems to *that* code) and allows you to use Delta's tabs.

There is one more way to sneak problem codes past the Apple II's operating system and that's to poke the codes directly to the output port. To send ASCII code 9, for example, you could do this:

```
100 N = 9
110 IF PEEK(49601)>127 THEN 110
120 POKE 49296,N
```

Line 110 checks the printer's status, and when it's ok, line 120 pokes the code to the printer.

### **Listing programs**

To make a listing of your BASIC programs on Delta from your Apple II computer you must take the following steps:

1. Be sure that the program that you wish to list is in the memory of the Apple II.
2. Direct the output to the printer by typing PR#1.
3. Type LIST to start the listing.
4. When the listing is finished, type PR#0 to redirect the output to the screen.

### **Program listings**

Following are program listings in Applesoft BASIC for the main programs used in the tutorial section of this book. The only



modifications that you might have to make are to the PR#1 and PR#0 commands as discussed above.

### Chart program

```

100 PR# 1
110 PRINT CHR$ (9);"255N"
120 GOSUB 1000: GOSUB 2000
130 PRINT "*REGULAR*"
140 GOSUB 3000
150 PRINT "*DOUBLE STRIKE*"
160 PRINT DS$;
170 GOSUB 3000
180 PRINT "*EMPHASIZED*"
190 EM = TRUE
200 GOSUB 3000
210 PRINT "*DOUBLE STRIKE & EMPHASIZED*"
220 PRINT DS$;EM$;
230 GOSUB 3000
240 END

1000 REM
1010 IT$ = CHR$ (27) + CHR$ (52)
1020 RO$ = CHR$ (27) + CHR$ (53)
1030 EN$ = CHR$ (27) + "W" + CHR$ (1)
1040 CE$ = CHR$ (27) + "W" + CHR$ (0)
1050 PI$ = CHR$ (27) + "B" + CHR$ (1)
1060 EL$ = CHR$ (27) + "B" + CHR$ (2)
1070 CO$ = CHR$ (27) + "B" + CHR$ (3)
1080 EM$ = CHR$ (27) + "E"
1090 CM$ = CHR$ (27) + "F"
1100 DS$ = CHR$ (27) + "G"
1110 CD$ = CHR$ (27) + "H"
1120 UN$ = CHR$ (27) + "-" + CHR$ (1)
1130 CU$ = CHR$ (27) + "-" + CHR$ (0)
1140 SP$ = CHR$ (27) + "S" + CHR$ (0)
1150 SB$ = CHR$ (27) + "S" + CHR$ (1)
1160 CS$ = CHR$ (27) + "T"
1170 RA$ = CHR$ (27) + "@"
1180 TRUE = 1:FALSE = 0
1190 RETURN
2000 REM
2050 PRINT RA$
2060 PRINT EN$;"  NORMAL      ENLARGED  "
2070 PRINT RA$;
2080 PRINT UN$;
2090 PRINT CO$;"CONDENSED ";

```



```

2100 PRINT EL$;" ELITE ";
2110 PRINT PI$;" PICA ";
2120 PRINT CO$;"CONDENSED ";
2130 PRINT EL$;" ELITE ";
2140 PRINT PI$;" PICA ";
2150 PRINT RA$
2160 RETURN
3000 REM
3050 IT = FALSE: PRINT RO$;
3060 UN = FALSE: PRINT CU$;
3070 EN = FALSE: PRINT CE$;
3080 PI = FALSE:
3100 PRINT CO$;
3110 GOSUB 3500:
3120 PRINT EL$;
3130 GOSUB 3500
3140 PRINT PI$;:PI = TRUE
3150 GOSUB 3500
3170 IF EN = TRUE THEN PRINT : GOTO 3190
3180 PRINT EN$;:EN = TRUE: GOTO 3080
3190 IF UN = TRUE THEN PRINT : GOTO 3210
3200 PRINT UN$;:UN = TRUE: GOTO 3070
3210 IF IT = TRUE THEN PRINT RA$: RETURN
3220 PRINT IT$;:IT = TRUE: GOTO 3060
3500 REM
3550 BL$=" " : FD$="...."
3560 IF EM = FALSE THEN PRINT "AB" + CHR$ (99) +
CHR$ (100);: GOTO 3610
3570 IF PI = FALSE THEN PRINT FD$;: GOTO 3590
3580 PRINT EM$;"AB"; CHR$ (99) + CHR$ (100);
3590 IF EN = TRUE THEN PRINT " " : RETURN
3600 IF EN = FALSE THEN PRINT BL$;: RETURN
3610 REM
3620 IF EN = TRUE THEN PRINT " " : RETURN
3630 PRINT SP$;"X" + CHR$ (120);
3640 PRINT SB$;"Y" + CHR$ (121) + " ";
3650 PRINT CS$;
3660 RETURN

```

### Special character chart program

```

10 L$ = CHR$ (27) + ">"
20 Z$ = CHR$ (27) + "="
25 PR# 1
27 PRINT CHR$ (9);"255N"
30 FOR J = 160 TO 255 STEP 8

```



```

40 FOR I = J TO J + 7
50 PRINT I;"=" ;L$; CHR$(I);Z$;" ";
60 NEXT I: PRINT : NEXT J
70 PR# 0
80 END

```

### Macro program

```

10 PR# 1
15 PRINT CHR$(9);"255N"
20 PRINT CHR$(27);"+";
30 PRINT CHR$(18);
40 PRINT CHR$(27);"W"; CHR$(0);
50 PRINT CHR$(27);"F";
60 PRINT CHR$(27);"H";
70 PRINT CHR$(27);"-"; CHR$(0);
80 PRINT CHR$(27);"T";
90 PRINT CHR$(27);"5";
100 PRINT CHR$(30)
110 PR# 0
120 END

```

### Bridge hand program

```

10 HOME
20 GOSUB 1000
30 GOSUB 2000
40 GOSUB 3000
50 GOSUB 4000
60 END
1000 REM
1010 DIM HA(4),DE(52),CA$(13),SU$(3)
1020 CA$(1) = " 2":CA$(2) = " 3":CA$(3) = " 4":
    CA$(4) = " 5" : CA$(5) = " 6"
1030 CA$(6) = " 7":CA$(7) = " 8":CA$(8) = " 9":
    CA$(9) = " 10"
1040 CA$(10) = " J":CA$(11) = " Q":CA$(12) = " K":
    CA$(13) = " A"
1050 SU$(0) = "S":SU$(1) = "H":SU$(2) = "D":
    SU$(3) = "C"
1080 RETURN
2000 REM
2010 PR# 1
2015 PRINT CHR$(9); CHR$(25)
2016 PRINT CHR$(25);"255N"

```



```
2020 PRINT CHR$ (27); CHR$ (68); CHR$ (20);
      CHR$ (40); CHR$ (0)
2030 PRINT CHR$ (27); CHR$ (43); CHR$ (27);
      CHR$ (36); CHR$ (0); CHR$ (27); CHR$ (69);
      CHR$ (30)
2040 PRINT CHR$ (27); CHR$ (42); CHR$ (0)
2045 FOR I = 1 TO 4
2050 PRINT CHR$ (27); CHR$ (42); CHR$ (1);
2060 FOR J = 1 TO 13
2070 READ X: PRINT CHR$ (X);
2080 NEXT J
2090 NEXT I
2100 PRINT
2110 RETURN
2120 DATA 72,11,4,10,20,10,52,72,52,10,20,10,4
2130 DATA 83,11,16,8,20,8,86,41,86,8,20,8,16
2140 DATA 67,11,8,16,8,18,65,62,65,18,8,16,8
2150 DATA 68,11,8,0,28,0,62,65,62,0,28,0,8
3000 REM
3010 FOR CA = 1 TO 52
3020 X = INT ( RND (1) * 4 + 1)
3030 IF HA(X) = 13 THEN 3020:
3035 HA(X) = HA(X) + 1
3040 DE(CA) = X
3050 NEXT CA
3060 RETURN
4000 REM
4010 PR# 1
4012 PRINT CHR$ (9); CHR$ (25)
4014 PRINT CHR$ (25); "255N":
4015 PRINT CHR$ (27); "!"; CHR$ (9); "NORTH"
4020 PRINT CHR$ (27); "$"; CHR$ (1); CHR$ (27);
      CHR$ (70);
4030 HA = 1
4040 FOR SU = 0 TO 3
4050 PRINT CHR$ (9);
4060 GOSUB 4300
4070 PRINT
4080 NEXT SU
4090 PRINT CHR$ (27); "!"; "WEST"; CHR$ (9);
      CHR$ (9); "EAST"
4100 PRINT CHR$ (27); "$"; CHR$ (1); CHR$ (27);
      CHR$ (70);
4110 FOR SU = 0 TO 3
```



```
4120 HA = 2
4130 GOSUB 4300
4140 PRINT CHR$ (9); CHR$ (9);
4150 HA = 3
4160 GOSUB 4300
4170 PRINT
4180 NEXT SU
4190 PRINT CHR$ (27);"!"; CHR$ (9);"SOUTH"
4200 PRINT CHR$ (27);"$"; CHR$ (1); CHR$ (27);
    CHR$ (70);
4210 HA = 4
4220 FOR SU = 0 TO 3
4230 PRINT CHR$ (9);
4240 GOSUB 4300
4250 PRINT
4260 NEXT SU
4270 PRINT CHR$ (27);"$"; CHR$ (0); CHR$ (27);
    CHR$ (70)
4280 RETURN
4300 PRINT SU$(SU);
4310 FOR CA = 13 TO 1 STEP - 1
4320 H1 = HA: IF DE(SU * 13 + CA) = H1 THEN PRINT
    CA$(CA);
4330 NEXT CA
4340 RETURN
```

### **Numeral program**

```
10 REM
20 PR# 1
25 PRINT CHR$ (9);"255N"
27 PRINT CHR$ (9); CHR$ (25)
30 DC$ = CHR$ (27) + CHR$ (42) + CHR$ (1)
40 DP$ = CHR$ (27) + CHR$ (88) + CHR$ (1)
50 CP$ = CHR$ (27) + CHR$ (88) + CHR$ (0)
60 L7$ = CHR$ (27) + "1":L12$ = CHR$ (27) +
    CHR$ (50)
70 FOR N1 = 33 TO 73
80 PRINT DC$;
90 PRINT CHR$ (N1);
100 READ N2
110 PRINT CHR$ (N2);
120 FOR S = 1 TO 11
130 READ MS
140 PRINT CHR$ (MS);
150 NEXT S
```



```
160 NEXT N1
180 AS = 33
190 FOR NU = 0 TO 9
210 NT$(NU) = CHR$(AS + 0) + CHR$(AS + 1)
220 NB$(NU) = CHR$(AS + 2) + CHR$(AS + 3)
225 AS = AS + 4
230 NEXT NU
240 BL$ = CHR$(73)
250 PRINT DP$;L7$
260 FOR NU = 0 TO 9
270 PRINT NT$(NU);BL$;
280 NEXT NU
290 PRINT
300 FOR NU = 0 TO 9
310 PRINT NB$(NU);BL$;
320 NEXT NU
330 PRINT CP$;L12$
340 REM ZERO
350 DATA 11,0,96,16,104,16,44,30,14,0,2,1
360 DATA 11,2,1,2,1,6,8,38,88,32,88,32
370 DATA 11,3,12,19,12,51,0,96,0,96,0,96
380 DATA 11,0,32,0,48,0,28,3,12,3,4,3
390 REM ONE
400 DATA 11,0,0,0,0,0,4,0,4,0,4,126
410 DATA 9,12,114,12,114,12,2,0,0,0,0,0
420 DATA 11,64,0,64,0,64,0,64,32,80,47,80
430 DATA 9,47,80,47,64,0,64,0,64,0,0,0
440 REM TWO
450 DATA 11,0,0,0,0,0,12,16,14,0,6,0
460 DATA 11,3,0,3,0,70,56,70,56,4,24,0
470 DATA 11,64,0,64,32,64,32,80,32,80,40,64
480 DATA 11,44,64,38,65,34,65,32,80,32,88,0
490 REM THREE
500 DATA 11,0,0,0,0,0,0,4,2,4,2,4
510 DATA 11,34,84,34,92,34,76,34,68,2,64,0
520 DATA 11,16,0,48,0,56,64,48,64,32,64,32
530 DATA 11,64,32,64,48,9,54,9,22,9,6,1
540 REM FOUR
550 DATA 11,0,0,0,0,0,0,64,36,88,32,16
560 DATA 11,0,0,64,32,64,56,64,60,2,12,0
570 DATA 11,0,8,4,10,5,10,5,8,4,72,4
580 DATA 11,88,38,89,38,89,6,73,4,8,6,0
590 REM FIVE
600 DATA 11,0,0,0,0,64,32,84,50,76,34,68
610 DATA 10,34,68,34,68,34,68,2,68,2,0,0
```



```

620 DATA 10,0,32,24,101,24,97,0,64,0,64,0
630 DATA 11,64,0,96,1,48,15,48,15,16,15,0
640 REM SIX
650 DATA 11,0,96,0,112,0,120,0,92,0,102,0
660 DATA 11,98,0,98,0,98,0,70,0,14,0,6
670 DATA 11,6,8,23,8,55,8,99,0,65,0,64
680 DATA 11,0,96,0,112,1,62,1,30,1,14,0
690 REM SEVEN
700 DATA 11,0,16,8,6,8,6,8,6,8,6,8
710 DATA 9,70,8,102,8,54,8,6,0,2,0,0
720 DATA 11,0,64,0,96,0,120,0,124,0,30,1
730 DATA 9,6,1,0,0,0,0,0,0,0,0,0
740 REM EIGHT
750 DATA 11,0,0,0,0,24,36,24,102,24,102,0
760 DATA 11,67,0,67,0,99,28,34,28,34,28,0
770 DATA 11,12,18,44,19,108,19,96,1,64,0,64
780 DATA 11,0,96,1,112,15,48,15,16,14,0,0
790 REM NINE
800 DATA 11,0,0,120,4,120,6,120,6,0,3,0
810 DATA 11,3,0,3,0,67,4,123,4,122,4,120
820 DATA 11,48,0,56,0,113,0,99,0,99,0,99
830 DATA 11,0,115,0,57,0,31,0,15,0,6,0
840 REM SPACE
850 DATA 11,0,0,0,0,0,0,0,0,0,0,0

```

### Download utility program

```

10 DIM Z(8,12),MM(11)
15 CS$ = "*" : SC$ = "@" : ST$ =
   " "
20 HOME : GOSUB 660
30 VTAB 24 : HTAB 34
40 GET A$
50 IF A$ = "P" THEN GOSUB 680 : GOTO 30
60 IF A$ = "E" THEN HOME : GOSUB 900 : GOSUB 260 :
   GOTO 30
70 IF A$ = CHR$(27) THEN HOME : END
80 PRINT CHR$(7) : GOTO 30
260 REM
265 FOR I = 1 TO 11 : MM(I) = 0 : NEXT I
270 VTAB 3 : HTAB 6 : PRINT CS$ ;
280 GET A$
290 IF A$ = "J" THEN GOSUB 390 : GOTO 370
300 IF A$ = "K" THEN GOSUB 410 : GOTO 370
310 IF A$ = "M" THEN GOSUB 430 : GOTO 370

```



```

320 IF A$ = "I" THEN GOSUB 450: GOTO 370
330 IF A$ = CHR$ (13) THEN GOSUB 470: GOTO 370
340 IF A$ = CHR$ (32) THEN GOSUB 490: GOTO 370
350 IF A$ = CHR$ (27) THEN GOSUB 500: GOTO 380
370 GOTO 280
380 RETURN
390 GOSUB 1000: Y = Y - 2: H = H - 1: IF Y < 1 THEN
PRINT CHR$ (7);: Y = 1: H = 1
400 GOSUB 1050: RETURN
410 GOSUB 1000: Y = Y + 2: H = H + 1: IF Y > 21 THEN
PRINT CHR$ (7);: Y = 21: H = 11
420 GOSUB 1050: RETURN
430 GOSUB 1000: X = X + 2: G = G + 1: IF X > 13 THEN
PRINT CHR$ (7);: X = 13: G = 7
440 GOSUB 1050: RETURN
450 GOSUB 1000: X = X - 2: G = G - 1: IF X < 1 THEN
PRINT CHR$ (7);: X = 1: G = 1
460 GOSUB 1050: RETURN
470 IF Z(G,H - 1) = 1 OR Z(G,H + 1) = 1 THEN PRINT
CHR$ (7);: RETURN
480 Z(G,H) = 1: INVERSE : VTAB X + 2: HTAB Y + 5:
PRINT SC$;: NORMAL : RETURN
490 Z(G,H) = 0: NORMAL : VTAB X + 2: HTAB Y + 5:
PRINT CS$;: RETURN
500 REM
510 FOR I = 2 TO 12: VTAB I: HTAB 29: PRINT
" ";: NEXT I
520 IF Z(G,H) = 1 THEN VTAB X + 2: HTAB Y + 5:
NORMAL : PRINT SC$;: GOTO 540
530 IF Z(G,H) = 0 THEN VTAB X + 2: HTAB Y + 5:
NORMAL : PRINT " ";
540 REM
550 FOR I = 1 TO 11: FOR J = 1 TO 7
560 MM(I) = MM(I) + Z(J,I) * 2 ^ (J - 1): NEXT J:
NEXT I
567 FOR I = 1 TO 11: S$ = STR$ (MM(I)): FOR J = 1 TO
LEN (S$)
568 VTAB J + 16: HTAB I * 2 + 4: PRINT
MID$ (S$,J,1);: NEXT J: NEXT I
580 GOSUB 660: RETURN
660 FOR I = 1 TO 7: FOR J = 1 TO 11: Z(I,J) = 0:
NEXT J: NEXT I
670 VTAB 24: HTAB 2: PRINT "E)DIT P)RINTER <ESC>
END ";: RETURN
680 REM

```



```

690 VTAB 21: HTAB 2: INPUT "NORMAL OR PROPORTIONAL
    (N/P) --> ";AN$
700 IF AN$ = "N" THEN PR = 0: GOTO 750
710 IF AN$ = "P" THEN 730
720 PRINT CHR$ (7);: GOTO 690
730 VTAB 21: HTAB 2: PRINT ST$;: VTAB 21: HTAB 2:
735 INPUT "PROPORTIONAL DATA (4-11) --> ";PR
740 IF PR < 4 OR PR > 11 THEN 730
750 VTAB 21: HTAB 2: PRINT ST$;: VTAB 21: HTAB 2:
755 INPUT "SHIFTED DOWN 1 ELSE ENTER 0 --> ";SH
760 IF SH < 0 OR SH > 1 THEN PRINT CHR$ (7);: GOTO
    750
770 VTAB 21: HTAB 2: PRINT ST$;: VTAB 21: HTAB 2:
775 INPUT "ENTER YOUR ASCII (33-126) --> ";AS
780 IF AS < 33 OR AS > 126 THEN 770
785 VTAB 21: HTAB 2: PRINT ST$;: VTAB 23: HTAB 38
790 IF SH = 1 THEN SH = 16
800 N1 = AS:N2 = PR + SH
810 FOR I = 1 TO 11:MM$ = MM$ + CHR$ (MM(I)):
    NEXT I
815 PR# 1
816 PRINT CHR$ (9);"255N"
820 PRINT CHR$ (27);"*"; CHR$ (1); CHR$ (N1);
    CHR$ (N2);MM$
825 IF AN$ = "N" THEN PRINT CHR$ (27);"$";
    CHR$ (1): GOTO 830
827 PRINT CHR$ (27);"X"; CHR$ (1)
830 FOR I = 1 TO 20: PRINT CHR$ (N1);" ";: NEXT I:
    PRINT
840 PRINT CHR$ (14);: FOR I = 1 TO 10: PRINT
    CHR$ (N1);" ";: NEXT I: PRINT
850 PRINT CHR$ (15);: FOR I = 1 TO 20: PRINT
    CHR$ (N1);" ";: NEXT I: PRINT CHR$ (18)
860 IF AN$ = "N" THEN PRINT CHR$ (27);"$"
    CHR$ (0): GOTO 870
865 PRINT CHR$ (27);"X"; CHR$ (0)
870 PRINT CHR$ (27);"@";:MM$ = "": RETURN
900 X = 1:Y = 1:G = 1:H = 1
901 HOME
902 FOR I = 2 TO 16 STEP 2: VTAB I: HTAB 5: FOR J =
    1 TO 23: PRINT "-";: NEXT J: PRINT : NEXT I
904 FOR J = 3 TO 16 STEP 2: VTAB J: FOR I = 5 TO 27
    STEP 2: HTAB I: PRINT "!";: NEXT I: PRINT :
    NEXT J
905 K = 1: VTAB 1: HTAB 5

```



```

906 FOR K = 1 TO 11: PRINT K;" ";: NEXT K
907 K = 0
908 FOR V = 3 TO 15 STEP 2: VTAB V: HTAB 2:
    PRINT 2 ^ K:K = K + 1: NEXT V
910 VTAB 3: HTAB 30: PRINT "CURSER "
912 VTAB 4: HTAB 29: PRINT "MOVEMENT"
913 VTAB 5: HTAB 29: PRINT "-----"
914 VTAB 6: HTAB 29: PRINT "<I> UP "
916 VTAB 7: HTAB 29: PRINT "<M> DOWN"
918 VTAB 8: HTAB 29: PRINT "<J> LEFT"
920 VTAB 9: HTAB 29: PRINT "<K> RIGHT"
922 VTAB 10: HTAB 29: PRINT "<RET> INSERT"
924 VTAB 11: HTAB 29: PRINT "<SPACE> DEL"
925 VTAB 12: HTAB 29: PRINT "<ESC> EXIT"
926 RETURN
1000 IF Z(G,H) = 0 THEN VTAB X + 2: HTAB Y + 5:
    PRINT " ";
1010 IF Z(G,H) = 1 THEN VTAB X + 2: HTAB Y + 5:
    PRINT SC$;
1020 RETURN
1050 IF Z(G,H) = 1 THEN INVERSE : VTAB X + 2:
    HTAB Y + 5: PRINT CS$;: NORMAL
1060 IF Z(G,H) = 0 THEN NORMAL : VTAB X + 2:
    HTAB Y + 5: PRINT CS$;: NORMAL
1070 RETURN

```

### Delta Plot program

```

4 HOME : PRINT " ": PRINT " "
5 PRINT " ": PRINT " "
6 PRINT "THIS PROGRAM TAKES ABOUT"
7 PRINT "1 MINUTE TO RUN. PLEASE"
8 PRINT " TURN ON YOUR PRINTER AND"
9 PRINT "STAND BY....."
10 A = 24576
20 FOR I = A TO A + 12
30 READ B
35 POKE I,B
40 NEXT I
50 DATA 32,74,255,165,250,5,251
60 DATA 133,252,32,63,255,96
100 REM DELTA-PLOT
110 DIM BIT%(75,14)
1000 REM SET PROGRAM CONSTANTS
1010 MASK%(1) = 64:MASK%(4) = 8
1020 MASK%(2) = 32:MASK%(5) = 4

```



```
1030 MASK%(3) = 16:MASK%(6) = 2
1040 LX = 20:LY = 20
1050 XFAC = 72 / LX:YFAC = 87 / LY
2000 REM PLOT CURVE
2010 RAD = 9
2020 X1 = 19:Y1 = 10
2030 FOR ANG = 0 TO 360 STEP 10
2040 R1 = ANG * 6.28 / 360
2050 X2 = RAD * COS (R1) + 10:Y2 = RAD * SIN (R1)
      + 10
2060 GOSUB 4000
2070 NEXT ANG
3000 REM SEND BIT IMAGE MAP TO PRINTER
3005 PR# 1
3006 PRINT CHR$(9);"0N"
3010 PRINT CHR$(27);"A"; CHR$(6)
3020 FOR ROW = 0 TO 14
3022 A$ = ""
3050 FOR COL = 1 TO 75
3060 A$ = A$ + CHR$(BIT%(COL,ROW))
3070 NEXT COL
3080 PRINT CHR$(27);"K"; CHR$(75); CHR$(0);A$
3090 NEXT ROW
3100 PRINT
3102 PR# 0
3230 END
4000 REM DRAW A LINE FROM X1,Y1 TO X2,Y2
4010 XL = X2 - X1:YL = Y2 - Y1
4020 NX = ABS (XL * XFAC):NY = ABS (YL * YFAC)
4030 IF NX < NY THEN NX = NY
4040 NS% = INT (NX + 1)
4050 DX = XL / NS%DY = YL / NS%
4060 FOR I = 1 TO NS%
4070 X1 = X1 + DX:Y1 = Y1 + DY
4080 GOSUB 5000
4090 NEXT I
4100 RETURN
5000 REM PLOT A POINT AT X1,Y1
5010 XX = X1 * XFAC:YY = Y1 * YFAC
5020 COL = INT (XX) + 1
5030 ROW = INT (YY / 6)
5040 XIT% = INT (YY - (6 * ROW)) + 1
5042 POKE 250,BIT%(COL,ROW)
5044 POKE 251,MASK%(XIT%)
5046 CALL 24576
```



```

5050 BIT%(COL,ROW) = PEEK (252)
5060 RETURN

```

### Pie chart program

```

4 HOME
5 PRINT " PLEASE STAND BY"
10 A = 768
20 FOR I = A TO A + 12
30 READ B
35 POKE I,B
40 NEXT I
50 DATA 32,74,255,165,250,5,251
60 DATA 133,252,32,63,255,96
100 REM PIECHART
110 DIM BIT%(190,36),A$(36),PCT%(25),TXT$(42),
    PTXT$(25)
120 ES$ = CHR$ (27):LF$ = CHR$ (10)
130 FF$ = CHR$ (12):VT$ = CHR$ (11)
140 EM$ = ES$ + "E":CE$ = ES$ + "F"
150 FOR I = 1 TO 168:SP$ = SP$ + CHR$ (0): NEXT I
1000 REM SET PROGRAM CONSTANTS
1020 MASK%(1) = 64:MASK%(4) = 8
1020 MASK%(2) = 32:MASK%(5) = 4
1030 MASK%(3) = 16:MASK%(6) = 2
1040 LX = 20:LY = 20
1050 XFAC = 190 / LX:YFAC = 216 / LY
1060 FOR I = 1 TO 42
1070 FOR J = 1 TO 80:TXT$(I) = TXT$(I) + " "
1080 NEXT J: NEXT I
1090 GOSUB 7000
1092 HOME : PRINT " ": PRINT " "
1094 PRINT " ": PRINT " "
1096 PRINT "THIS PROGRAM TAKES ABOUT"
1097 PRINT "4 MINUTES TO RUN. PLEASE"
1098 PRINT "TURN ON YOUR PRINTER AND"
1099 PRINT "STAND BY....."
2000 REM PLOT CURVE
2010 RAD = 9
2020 X1 = 19:Y1 = 10
2025 PRINT " ";
2030 FOR ANG = 0 TO 360 STEP 5
2040 R1 = ANG * 6.28 / 360
2050 X2 = RAD * COS (R1) + 10:Y2 = RAD * SIN (R1) +
10

```



```

2060 GOSUB 4000
2070 NEXT ANG
2080 FOR PI = 1 TO NP%
2090 X1 = 10:Y1 = 10
2100 TP% = TP% + PCT%(PI)
2110 ANG = 360 * TP% * .01
2120 R1 = ANG * 6.28 / 360
2130 X2 = RAD * COS (R1) + 10:Y2 = RAD * SIN (R1)
      + 10
2140 GOSUB 4000
2150 GOSUB 6000
2160 NEXT PI
3000 REM SEND BIT IMAGE MAP TO PRINTER
3020 FOR ROW = 0 TO 35
3022 A$(ROW) = ""
3050 FOR COL = 1 TO 190
3060 A$(ROW) = A$(ROW) + CHR$ (BIT%(COL,ROW))
3070 NEXT COL
3080 NEXT ROW
3090 PR# 1
3100 PRINT CHR$ (9);"0N"
3110 X = (40 - LEN (T1$) / 2)
3120 FOR I = 1 TO X: PRINT " ";: NEXT I
3130 PRINT EM$:TI$:CE$:LF$
3140 PRINT VT$:VT$:VT$
3150 PRINT ES$;"A"; CHR$ (3)
3160 PRINT TXT$(1);LF$;TXT$(2);LF$;TXT$(3);LF$
3170 FOR ROW = 0 TO 35
3180 PRINT ES$;"K"; CHR$ (102); CHR$ (1);
      SP$;A$(ROW)
3200 PRINT TXT$(ROW + 4)
3210 NEXT ROW
3220 PRINT TXT$(40);LF$
3230 PRINT TXT$(41);LF$
3240 PRINT TXT$(42);LF$
3250 PRINT ES$;"2";FF$
3255 PR# 0
3257 HOME
3260 END
4000 REM DRAW A LINE FROM X1,Y1, TO X2,Y2
4010 XL = X2 - X1:YL = Y2 - Y1
4020 NX = ABS (XL * XFAC):NY = ABS (YL * YFAC)
4030 IF NX < NY THEN NX = NY
4040 NS% = INT (NX + 1)
4050 DX = XL / NS%:DY = YL / NS%

```



```
4060 FOR I = 1 TO NS%
4070 X1 = X1 + DX:Y1 = Y1 + DY
4080 GOSUB 5000
4090 NEXT I
4100 RETURN
5000 REM PLOT A POINT AT X1,Y1
5010 XX = X1 * XFAC:YY = Y1 * YFAC
5020 COL = INT (XX) + 1
5030 ROW = INT (YY / 6)
5040 XIT% = INT (YY - (6 * ROW)) + 1
5042 POKE 250,BIT%(COL,ROW)
5044 POKE 251,MASK%(XIT%)
5046 CALL 768
5050 BIT%(COL,ROW) = PEEK (252)
5060 RETURN
6000 REM
6010 MA% = (ANG + PA%) / 2
6020 R1 = MA% * 6.28 / 360
6030 X3 = INT (20 * SIN (R1)):Y3 =
    INT (22 * COS (R1))
6040 X4 = 22 + X3:Y4 = 40 + Y3
6050 IF MA% > 270 OR MA% < 90 THEN GOSUB 6100:
    GOTO 6070
6060 GOSUB 6200
6070 PA% = ANG
6080 RETURN
6100 MM$ = TXT$(X4)
6102 LL$ = LEFT$ (MM$,Y4)
6104 PP = LEN (PTXT$(PI))
6106 RR$ = RIGHT$ (MM$,80 - (Y4 + PP))
6108 TXT$(X4) = LL$ + PTXT$(PI) + RR$
6110 RETURN
6200 MM$ = TXT$(X4)
6202 PP = LEN (PTXT$(PI))
6204 LL$ = LEFT$ (MM$,(Y4 - PP))
6206 RR$ = RIGHT$ (MM$,(80 - Y4))
6208 TXT$(X4) = LL$ + PTXT$(PI) + RR$
6210 RETURN
7000 REM
7005 I = 1
7010 HOME : PRINT : PRINT : PRINT
7020 INPUT "ENTER TITLE FOR CHART ";TI$
7030 AS% = 0:AL% = 100
7040 HOME
7050 PRINT "TOTAL SO FAR      : ";AS%
```



```
7060 PRINT "TOTAL REMAINING : ";AL%
7070 INPUT "ENTER % FOR FIELD ";PCT%(I)
7080 IF PCT%(I) > AL% OR PCT%(I) = 0 THEN
    PCT%(I) = AL%
7090 AL% = AL% - PCT%(I)
7100 AS% = AS% + PCT%(I)
7110 INPUT "ENTER DESCRIPTION OF FIELD : ";PTXT$(I)
7120 IF LEN (PTXT$(I)) > 15 THEN PRINT "FIELD TOO
    LONG - 15 CHAR. MAX": GOTO 7110
7130 IF AL% = 0 THEN GOTO 7200
7140 I = I + 1
7150 GOTO 7040
7200 NP% = I
7210 IF NP% = 1 THEN 7040
7220 HOME
7230 RETURN
```



## Appendix D

# TRS-80 Computers

All that's required to connect Delta to your TRS-80 is a cable. It is available at your Delta dealer.

When connecting Delta to a TRS-80 we recommend that you set the DIP switches in Delta as shown below.

### Recommended DIP Switch Settings for TRS-80

| Switch | Setting | Function                  |
|--------|---------|---------------------------|
| 1-1    | ON      | 11 inch page size         |
| 1-2    | ON      | Normal print density      |
| 1-3    | ON      | 10 CPI pitch              |
| 1-4    | ON      | Normal characters         |
| 1-5    | ON      | 1/6 inch line feed        |
| 1-6    | ON      | U.S.A. Character set      |
| 1-7    | ON      |                           |
| 1-8    | ON      |                           |
| 2-1    | ON      | Paper-out detector active |
| 2-2    | OFF     | Parallel interface        |
| 2-3    | OFF     | 8-bit interface           |
| 2-4    | ON      | Auto line feed            |

### TRS-80 Model I Parallel Cable

| Delta   |          | TRS-80 Model I |          |
|---------|----------|----------------|----------|
| Pin No. | Function | Pin No.        | Function |
| 1       | STROBE   | 1              | STROBE   |
| 2       | D1       | 3              | D1       |
| 3       | D2       | 5              | D2       |
| 4       | D3       | 7              | D3       |
| 5       | D4       | 9              | D4       |
| 6       | D5       | 11             | D5       |
| 7       | D6       | 13             | D6       |
| 8       | D7       | 15             | D7       |
| 9       | D8       | 17             | D8       |
| 11      | BUSY     | 21             | READY    |



**TRS-80 Model II Parallel Cable**

| Delta   |          | TRS-80 Model II |          |
|---------|----------|-----------------|----------|
| Pin No. | Function | Pin No.         | Function |
| 1       | STROBE   | 1               | STROBE   |
| 2       | D1       | 3               | D1       |
| 3       | D2       | 5               | D2       |
| 4       | D3       | 7               | D3       |
| 5       | D4       | 9               | D4       |
| 6       | D5       | 11              | D5       |
| 7       | D6       | 13              | D6       |
| 8       | D7       | 15              | D7       |
| 9       | D8       | 17              | D8       |
| 10      | ACK      | 19              | ACK      |
| 11      | BUSY     | 21              | BUSY     |

**TRS-80 BASIC**

You may have to initialize your Model II to direct LPRINT statements to the printer. Use the SYSTEM "FORMS" command to do it.

TRS-80 uses another version of Microsoft Basic. Most of the programs in this book will work just as they are, but the TRS-80 does have a few unique "problem codes". They are 0, 10, 11, and 12. None of these are passed properly to the printer.

You can bypass the TRS-80's BASIC and send these codes directly to the printer with the following short routine. The variable N must be set equal to the code that you wish to pass (in our example it's 0).

```
90 N = 0
100 IF PEEK(14312) < > 63 THEN 100
110 POKE 14312, N
```

Or you can use this special printer driver that will solve all your problems. Just run this program first, and then any codes sent by a BASIC program will be sent directly to the printer. This program is for the TRS-80 Model III.

```
5 REM DRIVER FOR TRS-80 III
10 AD=16571
20 FOR I=0 TO 14
30 READ A:POKE AD+I,A
```



```
40 NEXT I
50 POKE 16422,187
60 POKE 16423,64
70 DATA 33,232,55,203,126,32,252,33,17,
    0,57,126,211,251,201
80 END
```

And here is a version for the TRS-80 Model I.

```
5 REM DRIVER FOR TRS-80 I
10 AD=16571
20 FOR I=0 TO 15
30 READ A:POKE AD+I,A
40 NEXT I
50 POKE 16422,187
60 POKE 16423,64
70 DATA 33,232,55,203,126,32,252,33,17,
    0,57,126,50,232,55,201
80 END
```

### Chart program

```
100 CLEAR 1000
110 GOSUB 1000
120 GOSUB 2000
130 LPRINT "*REGULAR*"
140 GOSUB 3000
150 LPRINT "*DOUBLE STRIKE*"
160 LPRINT DS$;
170 GOSUB 3000
180 LPRINT "*EMPHASIZED*"
190 EM = TRUE
200 GOSUB 3000
210 LPRINT "*DOUBLE STRIKE & EMPHASIZED*"
220 LPRINT DS$ EM$;
230 GOSUB 3000
240 END
1000 REM
1060 IT$ = CHR$(27) + CHR$(52)
1070 RO$ = CHR$(27) + CHR$(53)
1080 REM
1090 EN$ = CHR$(27) + CHR$(87) + CHR$(1)
1100 NW$ = CHR$(27) + CHR$(87) + CHR$(0)
1110 PI$ = CHR$(27) + CHR$(66) + CHR$(1)
```



```

1120 EL$      = CHR$(27) + CHR$(66) + CHR$(2)
1130 CO$ = CHR$(27) + CHR$(66) + CHR$(3)
1140 REM
1150 EM$      = CHR$(27) + CHR$(69)
1160 NE$      = CHR$(27) + CHR$(70)
1170 DS$ = CHR$(27) + CHR$(71)
1180 ND$ = CHR$(27) + CHR$(72)
1190 UN$      = CHR$(27) + CHR$(45) + CHR$(1)
1200 NU$      = CHR$(27) + CHR$(45) + CHR$(0)
1210 SP$      = CHR$(27) + CHR$(83) + CHR$(0)
1220 SB$      = CHR$(27) + CHR$(83) + CHR$(1)
1230 NS$      = CHR$(27) + CHR$(84)
1240 RA$ = NE$+NU$+ND$
1250 RA$ = RA$ + RO$ + PI$ +NW$
1260 REM
1270 TRUE = 1 : FALSE = 0
1290 RETURN
2000 REM
2050 LPRINT RA$
2060 LPRINT EN$ "    NORMAL        ENLARGED  "
2070 LPRINT RA$;
2080 LPRINT UN$;
2090 LPRINT CO$ "CONDENSED ";
2100 LPRINT EL$      " ELITE    ";
2110 LPRINT PI$      " PICA     ";
2120 LPRINT CO$ "CONDENSED ";
2130 LPRINT EL$      " ELITE    ";
2140 LPRINT PI$      " PICA     "
2150 LPRINT RA$
2160 RETURN
3000 REM
3050 IT      = FALSE           :LPRINT RO$;
3060 UN = FALSE           :LPRINT NU$;
3070 EN      = FALSE           :LPRINT NW$;
3080 PI      = FALSE
3090 REM
3100 LPRINT CO$;
3110 GOSUB 3500 'PRINT AS REQUIRED
3120 LPRINT EL$;
3130 GOSUB 3500 'PRINT AS REQUIRED
3140 LPRINT PI$; :PI = TRUE
3150 GOSUB 3500 'PRINT AS REQUIRED
3160 REM
3170 IF EN = TRUE THEN LPRINT :GOTO 3190
3180 LPRINT EN$; :EN = TRUE :GOTO 3080

```



```

3190 IF UN = TRUE THEN LPRINT :GOTO 3210
3200 LPRINT UN$; :UN = TRUE :GOTO 3070
3210 IF IT = TRUE THEN LPRINT RA$ :RETURN
3220 LPRINT IT$; :IT = TRUE :GOTO 3060
3500 REM
3550 BL$ = STRING$(6,32) :FD$ = "...."
3560 IF EM = FALSE THEN LPRINT "ABcd"; :GOTO 3610
3570 IF PI = FALSE THEN LPRINT FD$; :GOTO 3590
3580 LPRINT EM$ "ABcd" ;
3590 IF EN = TRUE THEN LPRINT " " ; :ELSE
      LPRINT BL$;
3600 RETURN
3610 REM
3620 IF EN = TRUE THEN LPRINT " " ; :RETURN
3630 LPRINT SP$; "Xx";
3640 LPRINT SB$; "Yy ";
3650 LPRINT NS$;
3660 RETURN

```

### Special character chart program

```

10 FOR J = 160 TO 255 STEP 8
20 FOR I = J TO J+ 7
30 LPRINT I "=" CHR$(I) CHR$(9);
40 NEXT I : LPRINT : NEXT J

```

### Macro program

```

10 LPRINT CHR$(27) "+"; ' START DEFINITION
   OF MACRO
20 LPRINT CHR$(18); ' PICA
30 LPRINT CHR$(27) "W" CHR$(0); ' EXPANDED OFF
40 LPRINT CHR$(27) "F"; ' EMPHASIZED OFF
50 LPRINT CHR$(27) "H"; ' DOUBLE-STRIKE OFF
60 LPRINT CHR$(27) "-" CHR$(0); ' UNDERLINE OFF
70 LPRINT CHR$(27); "T"; ' SUPER & SUBSCRIPT
   OFF
80 LPRINT CHR$(27); "5";
90 LPRINT CHR$(30); ' END MACRO DEFINITION

```

### Bridge hand program

```

20 GOSUB 1000
30 GOSUB 2000
40 GOSUB 3000
50 GOSUB 4000

```



```

70 END
1000 REM INITIALIZE VARIABLES
1010 DIM HA(4), DE(52), CA$(13), SU$(3)
1020 CA$(1)=" 2" : CA$(2)=" 3" : CA$(3)=" 4"
1030 CA$(4)=" 5" : CA$(5)=" 6" : CA$(6)=" 7"
1040 CA$(7)=" 8" : CA$(8)=" 9" : CA$(9)=" 10"
1050 CA$(10)=" J" : CA$(11)=" Q" : CA$(12)=" K" :
      CA$(13)=" A"
1060 SU$(0)="S" : SU$(1)="H" : SU$(2)="D" :
      SU$(3)="C"
1070 RETURN
2000 REM INITIALIZE PRINTER
2010 LPRINT CHR$(27); CHR$(68); CHR$(20); CHR$(40);
      CHR$(0)
2020 LPRINT CHR$(27) CHR$(43); CHR$(27) CHR$(36)
      CHR$(0); CHR$(27) CHR$(69) CHR$(30)
2030 LPRINT CHR$(27) CHR$(42) CHR$(0)
2040 FOR I=1 TO 4
2050 LPRINT CHR$(27); CHR$(42); CHR$(1);
2060 FOR J=1 TO 13
2070 READ X : LPRINT CHR$(X);
2080 NEXT J
2090 NEXT I
2100 LPRINT
2110 RETURN
2120 DATA 72,11,4,10,20,10,52,72,52,10,20,10,4
2130 DATA 83,11,16,8,20,8,86,41,86,8,20,8,16
2140 DATA 67,11,8,16,8,18,65,62,65,18,8,16,8
2150 DATA 68,11,8,0,28,0,62,65,62,0,28,0,8
3000 REM DEAL CARD
3010 FOR CA = 1 TO 52
3020 X = INT(RND(0) * 4 + 1)
3030 IF HA(X)=13 THEN 3020
3035 HA(X)=HA(X) + 1
3040 DE(CA)=X
3050 NEXT CA
3060 RETURN
4000 REM PRINT FOUR HANDS
4010 LPRINT CHR$(27); "!" ; CHR$(9); "NORTH"
4020 LPRINT CHR$(27); "$" ; CHR$(1); CHR$(27);
      CHR$(70);
4030 HA = 1
4040 FOR SU = 0 TO 3
4050 LPRINT CHR$(9);
4060 GOSUB 4300

```



```
4070 LPRINT
4080 NEXT SU
4090 LPRINT CHR$(27); "!"; "WEST"; CHR$(9); CHR$(9);
      "EAST"
4100 LPRINT CHR$(27); "$"; CHR$(1); CHR$(27);
      CHR$(70);
4110 FOR SU = 0 TO 3
4120 HA = 2
4130 GOSUB 4300
4140 LPRINT CHR$(9) CHR$(9);
4150 HA = 3
4160 GOSUB 4300
4170 LPRINT
4180 NEXT SU
4190 LPRINT CHR$(27); "!"; CHR$(9); "SOUTH"
4200 LPRINT CHR$(27); "$"; CHR$(1); CHR$(27);
      CHR$(70);
4210 HA = 4
4220 FOR SU = 0 TO 3
4230 LPRINT CHR$(9);
4240 GOSUB 4300
4250 LPRINT
4260 NEXT SU
4270 LPRINT CHR$(27); "$"; CHR$(0); CHR$(27);
      CHR$(70)
4280 RETURN
4290 REM PRINT ONE LINE
4300 LPRINT SU$(SU);
4310 FOR CA = 13 TO 1 STEP -1
4320 IF DE(SU*13+CA)=HA THEN LPRINT CA$(CA);
4330 NEXT CA
4340 RETURN
```

### **Numeral program**

```
10 REM PROGRAM TO DEFINE AND PRINT NUMERAL
20 REM EACH NUMERAL IS MADE UP OF 4 CHARACTERS (2
   WIDE X 2 HIGH)
30 DD$ = CHR$(27)+CHR$(42)+CHR$(1)
40 DP$ = CHR$(27)+CHR$(88)+CHR$(1)
50 ND$ = CHR$(27)+CHR$(88)+CHR$(0)
60 L7$ = CHR$(27)+CHR$(49) : L12$ =
   CHR$(27)+CHR$(50)
70 FOR N1 = 160 TO 200
80 LPRINT DD$;
```



```
90 LPRINT CHR$(N1);
100 READ N2
110 LPRINT CHR$(N2);
120 FOR S = 1 TO 11
130 READ MS
140 LPRINT CHR$(MS);
150 NEXT S
160 NEXT N1
180 AS = 160
190 FOR NUM = 0 TO 9
200 NT$(NUM) = CHR$(AS + 0) + CHR$(AS + 1)
210 NB$(NUM) = CHR$(AS + 2) + CHR$(AS + 3)
220 AS = AS + 4
230 NEXT NUM
240 BK$ = CHR$(200)
250 LPRINT DP$; L7$
260 FOR NUM = 0 TO 9
270 LPRINT NT$(NUM);BK$;
280 NEXT NUM
290 LPRINT
300 FOR NUM = 0 TO 9
310 LPRINT NB$(NUM);BK$;
320 NEXT NUM
330 LPRINT ND$; L12$
340 REM ZERO
350 DATA 11,0,96,16,104,16,44,30,14,0,2,1
360 DATA 11,2,1,2,1,6,8,38,88,32,88,32
370 DATA 11,3,14,19,14,51,0,96,0,96,0,96
380 DATA 11,0,32,0,48,0,28,3,14,3,4,3
390 REM ONE
400 DATA 11,0,0,0,0,0,4,0,4,0,4,126
410 DATA 9,14,114,14,114,14,2,0,0,0,0,0
420 DATA 11,64,0,64,0,64,0,64,32,80,47,80
430 DATA 9,47,80,47,64,0,64,0,64,0,0,0
440 REM TWO
450 DATA 11,0,0,0,0,0,14,16,14,0,6,0
460 DATA 11,3,0,3,0,70,56,70,56,4,24,0
470 DATA 11,64,0,64,32,64,32,80,32,80,40,64
480 DATA 11,44,64,38,65,34,65,32,80,32,88,0
490 REM THREE
500 DATA 11,0,0,0,0,0,0,4,2,4,2,4
510 DATA 11,34,84,34,92,34,76,34,68,2,64,0
520 DATA 11,16,0,48,0,56,64,48,64,32,64,32
530 DATA 11,64,32,64,48,9,54,9,22,9,6,1
540 REM FOUR
```



```

550 DATA 11,0,0,0,0,0,0,64,36,88,32,16
560 DATA 11,0,0,64,32,64,56,64,60,2,14,0
570 DATA 11,0,8,4,10,5,10,5,8,4,72,4
580 DATA 11,88,38,89,38,89,6,73,4,8,6,0
590 REM FIVE
600 DATA 11,0,0,0,0,64,32,84,50,76,34,68
610 DATA 10,34,68,34,68,34,68,2,68,2,0,0
620 DATA 10,0,32,24,101,24,97,0,64,0,64,0
630 DATA 11,64,0,96,1,48,15,48,15,16,15,0
640 REM SIX
650 DATA 11,0,96,0,112,0,120,0,92,0,102,0
660 DATA 11,98,0,98,0,98,0,70,0,14,0,6
670 DATA 11,7,8,23,8,55,8,99,0,65,0,64
680 DATA 11,0,96,0,112,1,62,1,30,1,14,0
690 REM SEVEN
700 DATA 11,0,16,8,6,8,6,8,6,8,6,8
710 DATA 9,70,8,102,8,54,8,6,0,2,0,0
720 DATA 11,0,64,0,96,0,120,0,124,0,30,1
730 DATA 9,6,1,0,0,0,0,0,0,0,0,0
740 REM EIGHT
750 DATA 11,0,0,0,0,24,36,24,102,24,102,0
760 DATA 11,67,0,67,0,99,28,34,28,34,28,0
770 DATA 11,14,18,44,19,108,19,96,1,64,0,64
780 DATA 11,0,96,1,112,15,48,15,16,14,0,0
790 REM NINE
800 DATA 11,0,0,120,4,120,6,120,6,0,3,0
810 DATA 11,3,0,3,0,67,4,123,4,122,4,120
820 DATA 11,48,0,56,0,113,0,99,0,99,0,99
830 DATA 11,0,115,0,57,0,31,0,15,0,7,0
840 REM SPACE
850 DATA 11,0,0,0,0,0,0,0,0,0,0,0

```

### Download utility program

```

4 CLEAR 1000
5 ME$="E)DIT P)RINT Q)UIT
                                     "
6 BL$=STRING$(63," ")
10 DIM Z(8,12),MM(11)
20 CLS:GOSUB 660
30 CS$="@":SC$=CHR$(143):SS$="*"
40 A$=INKEY$:IF A$="" THEN 40
50 IF A$="Q" THEN CLS:END
60 IF A$="P" THEN GOSUB 680:GOTO 40
70 IF A$="E" THEN CLS:GOSUB 90:GOSUB 260:GOTO 40
80 GOTO 40

```



```

90 X=1:Y=1:G=1:H=1:
100 FOR I=1 TO 11:MM(I)=0:NEXT I
105 CLS
120 FOR I=0 TO 7 : PRINT "      ";
125 FOR J=1 TO 11:PRINT "!---";:NEXT J:PRINT "!" :IF
    I<7 THEN PRINT:
126 NEXT I
130 FOR I=0 TO 6:PRINT @64*I+64*I+64,2^I;:NEXT I
140 PRINT @70,CS$;
150 PRINT @180,"<R> RIGHT";
160 PRINT @244,"<L> LEFT";
170 PRINT @308,"<U> UP";
180 PRINT @372,"<D> DOWN";
190 PRINT @436,"<I> INSERT";
200 PRINT @500,"<C> CLEAR";
210 PRINT @564,"<Q> QUIT";
220 RETURN
260 REM EDIT LEVEL
270 A$=INKEY$:IF A$="" THEN 270
280 IF A$="L" THEN GOSUB 390:GOTO 370
290 IF A$="R" THEN GOSUB 410:GOTO 370
300 IF A$="D" THEN GOSUB 430:GOTO 370
310 IF A$="U" THEN GOSUB 450:GOTO 370
320 IF A$="I" THEN GOSUB 470:GOTO 370
330 IF A$="C" THEN GOSUB 490:GOTO 370
340 IF A$="Q" THEN GOSUB 500:GOTO 380
370 GOTO 270
380 RETURN
390 GOSUB 920:Y=Y-4:H=H-1:IF Y<1 THEN Y=1:H=1
400 GOSUB 950:RETURN
410 GOSUB 920:Y=Y+4:H=H+1:IF Y>41 THEN Y=41:H=11
420 GOSUB 950:RETURN
430 GOSUB 920:X=X+2:G=G+1:IF X>13 THEN X=13:G=7
440 GOSUB 950:RETURN
450 GOSUB 920:X=X-2:G=G-1:IF X<1 THEN X=1:G=1
460 GOSUB 950:RETURN
470 IF Z(G,H-1)=1 OR Z(G,H+1)=1 THEN RETURN
480 Z(G,H)=1:PRINT @X*64+Y+5,SS$;:RETURN
490 Z(G,H)=0:PRINT @X*64+Y+5,CS$;:RETURN
500 REM GET OF EDIT MODE
510 IF Z(G,H)=1 THEN PRINT @X*64+Y+5,SC$;
520 IF Z(G,H)=0 THEN PRINT @X*64+Y+5," ";
524 PRINT @X*64+Y+5, " ";
530 FOR I=1 TO 11:FOR J=1 TO 7:
    MM(I)=MM(I)+Z(J,I)*2[(J-1):NEXT J:NEXT I

```



```
540 CLS
542 FOR I=1 TO 11:PRINT "M";I;TAB(5);"=" ;MM(I):
    NEXT I
550 GOSUB 660:RETURN
660 FOR I=1 TO 7:FOR J=1 TO 11:Z(I,J)=0:NEXT J:
    NEXT I
670 PRINT @960,ME$;:RETURN
680 REM PRINT MODE
690 PRINT @832,"";:INPUT "NORMAL OR PROPORTIONAL -->
    ";AN$
700 IF AN$="N" THEN PR=0:GOTO 750
710 IF AN$="P" THEN GOTO 730
720 GOSUB 1000:GOTO 690
730 GOSUB 1000:PRINT @832,"";:INPUT "PROPORTIONAL
    DATA (4-11) --> ";PR
740 IF PR<4 OR PR>11 THEN 730
750 GOSUB 1000:PRINT @832,"";:INPUT "IF SHIFTED
    ENTER 1 ELSE ENTER 0 --> ";SH
760 IF SH<0 OR SH>1 THEN 750
770 GOSUB 1000:PRINT @832,"";:INPUT "ASCII (33-126)
    (160-254) ";AS
780 IF (AS<33 OR AS>126) AND (AS<160 OR AS>254) THEN
    770
790 GOSUB 1000
800 IF SH=1 THEN SH=16 ELSE SH=0
810 N1=AS:N2=PR+SH
820 FOR I=1 TO 11:MM$=MM$+CHR$(MM(I)):NEXT I
830 LPRINT CHR$(27);"*";CHR$(1);CHR$(N1);CHR$(N2);
    MM$
840 IF AN$="N" THEN LPRINT CHR$(27);"$";CHR$(1):GOTO
    860
850 LPRINT CHR$(27);"X";CHR$(1)
860 FOR I=1 TO 20:LPRINT CHR$(N1);" ";:NEXT I:LPRINT
870 LPRINT CHR$(14);:FOR I=1 TO 10:LPRINT
    CHR$(N1);:NEXT I:LPRINT
880 LPRINT CHR$(15);:FOR I=1 TO 20:LPRINT
    CHR$(N1);:NEXT I:LPRINT
890 IF AN$="N" THEN LPRINT CHR$(27);"$";CHR$(0):GOTO
    896
894 LPRINT CHR$(27);"X";CHR$(0)
896 LPRINT CHR$(27);"@":MM$="":GOSUB 660:RETURN
920 IF Z(G,H)=0 THEN PRINT @X*64+Y+5," ";
930 IF Z(G,H)=1 THEN PRINT @X*64+Y+5,SC$;
940 RETURN
950 IF Z(G,H)=1 THEN PRINT @X*64+Y+5,SS$;
```



```

960 IF Z(G,H)=0 THEN PRINT @X*64+Y+5,CS$;
970 RETURN
1000 PRINT @832,BL$;:RETURN

```

### Delta Plot program

```

5 CLEAR 1000
10 CLS:PRINT "":PRINT "":PRINT ""
20 PRINT "THIS PROGRAM TAKES ABOUT 1 MINUTE TO RUN"
30 PRINT "PLEASE TURN ON YOUR PRINTER AND STAND BY"
100 DIM BIT%(76,14)
110 MASK%(1)=128      : MASK%(4)=16
120 MASK%(2)=64       : MASK%(5)=8
130 MASK%(3)=32       : MASK%(6)=4
140 LX=20      : LY=20
150 XFAC=72/LX   : YFAC=87/LY
1000 REM PLOT CURVE
1010 RAD=9
1020 X1=19      : Y1=10
1030 FOR ANG=0 TO 360 STEP 10
1040 R1=ANG*6.28/360
1050 X2=RAD*COS(R1)+10 : Y2=RAD*SIN(R1)+10
1060 GOSUB 3000
1070 NEXT ANG
2000 REM
2010 LPRINT CHR$(27) "A" CHR$(6)
2020 FOR ROW=0 TO 14
2030 A$=""
2040 LPRINT CHR$(27);"K";CHR$(75);CHR$(0);
2050 FOR COL=1 TO 75
2060 MM=(BIT%(COL,ROW)
2065 IF MM=12 THEN MM=140
2070 A$=A$ + CHR$(MM)
2090 NEXT COL
2100 LPRINT A$
2110 NEXT ROW
2120 LPRINT CHR$(27) "2"
2130 END
3000 REM DRAW A LINE FROM X1,Y1 TO X2,Y2
3010 XL=X2 - X1      : YL=Y2 - Y1
3020 NX=ABS(XL*XFAC) : NY=ABS(YL*YFAC)
3030 IF NX < NY THEN NX=NY
3040 NS%=INT(NX+1)
3050 DX=XL/NS%       : DY=YL/NS%
3060 FOR I=1 TO NS%
3070 X1=X1 + DX      : Y1=Y1 + DY

```



```

3080 GOSUB 4000
3090 NEXT I
3100 RETURN
4000 REM PLOT A POINT AT X1,Y1
4010 XX=X1 * XFAC      : YY=Y1 * YFAC
4020 COL=INT(XX) + 1
4030 ROW=INT(YY/6)
4040 XIT%=INT(YY-(6*ROW))+1
4050 BIT%(COL,ROW)=BIT%(COL,ROW) OR MASK%(XIT%)
4060 RETURN

```

### Pie chart program

```

40 CLS
45 PRINT "                PLEASE STAND BY"
50 CLEAR 10000
100 ' PIECHART
110 ESC$ = CHR$(27)      : LF$ = CHR$(10)
120 FF$ = CHR$(12)      : VT$ = CHR$(11)
130 EM$ = ESC$ + "E"    : NE$ = ESC$ + "F"
1000 ' Set program constants
1020 DIM BIT%(190,36),A$(36),PCT%(25),
      TXT$(42),PXT$(25)
1030 MASK%(1) = 128      : MASK%(4) = 16
1040 MASK%(2) = 64       : MASK%(5) = 8
1050 MASK%(3) = 32       : MASK%(6) = 4
1060 LX = 20 : LY = 20
1070 XFAC = 190/LX      : YFAC = 216/LY
1080 FOR I= 1 TO 42
1085 REM THERE ARE 80 SPACES IN TXT$(I)
1090 TXT$(I) = " (80 spaces) " :
1100 NEXT I
1110 GOSUB 7000
1120 CLS:PRINT " THIS PROGRAM TAKES ABOUT 5 MINUTES
      TO RUN"
1130 PRINT "SO TURN ON YOUR PRINTER AND STAND
      BY....."
2000 ' Plot curve
2010 RD = 9
2020 X1 = 19 : Y1 = 10
2040 FOR ANG% = 0 TO 360 STEP 5
2050 RANG = ANG%*6.28/360
2060 X2 = RD*COS(RANG)+10 : Y2 = RD*SIN(RANG)+10
2070 GOSUB 4000
2080 NEXT ANG%
2090 FOR PIE = 1 TO NP%

```



```

2100 X1 = 10 : Y1 = 10
2110 TP%=TP%+PCT%(PIE)
2120 ANG%=360*TP%*.01
2130 RANG = ANG%*6.28/360
2140 X2 = RD*COS(RANG)+10 : Y2 = RD*SIN(RANG)+10
2150 GOSUB 4000
2160 GOSUB 6000
2170 NEXT PIE
3000 ' SEND BIT IMAGE MAP TO PRINTER
3020 FOR ROW% = 0 TO 35
3030 A$(ROW%) = ""
3040 FOR COL% = 1 TO 190
3045 MM=BIT%(COL%,ROW%)
3047 IF MM=12 THEN MM=140
3050 A$(ROW%) = A$(ROW%) + CHR$(MM)
3060 NEXT COL%
3080 NEXT ROW%
3090 LL=LEN (TT$):NN=40-LL/2:FOR I=1 TO NN:LPRINT
    " ";:NEXT I
3091 LPRINT EM$;TI$;NE$;LF$
3100 LPRINT VT$;VT$;VT$
3110 LPRINT ESC$;"A";CHR$(3)
3120 LPRINT TXT$(1);LF$;TXT$(2);LF$;TXT$(3);LF$
3130 FOR ROW% = 0 TO 35
3140 LPRINT " ";
    ESC$;"K";CHR$(190);CHR$(0);
3150 LPRINT A$(ROW%)
3160 LPRINT TXT$(ROW%+4)
3180 NEXT ROW%
3190 LPRINT TXT$(40);LF$
3200 LPRINT TXT$(41);LF$
3210 LPRINT TEXT$(42);LF$
3220 LPRINT ESC$;"2";FF$
3230 END
4000 ' DRAW A LINE FROM X1,Y1 TO X2,Y2
4010 XL = X2 - X1 : YL = Y2 - Y1
4020 NX = ABS(XL*XFAC) : NY = ABS(YL*YFAC)
4030 IF NX < NY THEN NX = NY
4040 NS% = INT(NX+1)
4050 DX = XL/NS% : DY = YL/NS%
4060 FOR I% = 1 TO NS%
4070 X1 = X1 + DX : Y1 = Y1 + DY
4080 GOSUB 5000
4090 NEXT I%
4110 RETURN

```



```
5000 ' PLOT A POINT AT X1,Y1
5010 XX = X1 * XFAC      : YY = Y1 * YFAC
5020 COL% = INT(XX) + 1
5030 ROW% = INT(YY/6)
5040 XIT% = INT(YY - ROW% * 6)+1
5050 BIT%(COL%,ROW%) = BIT%(COL%,ROW%) OR
      MASK%(XIT%)
5060 RETURN
6000 ' Place text fields in the proper location
6010 MA%=(ANG%+PA%)/2
6020 RANG = MA%*6.28/360
6030 X3 = INT(20*SIN(RANG)) :
      Y3 = INT(22*COS(RANG))
6040 X4 = 22 + X3      : Y4 = 40 + Y3
6050 IF MA%>270 OR MA%<90 THEN MID$(TXT$(X4),Y4) =
      PXT$(PIE) ELSE MID$(TXT$(X4),Y4-
      LEN(PXT$(PIE)))=PXT$(PIE)
6060 PA%=ANG%
6070 RETURN
7000 ' ACCEPT DATA FROM SCREEN
7010 CLS: PRINT : PRINT : PRINT
7020 INPUT "ENTER TITLE FOR CHART: ";TT$
7030 AS%=0      : AL%=100
7040 I=1
7050 CLS: PRINT "      ENTER PARAMETERS FOR
      PIE-CHART"
7060 PRINT " TOTAL SO FAR : ";
7070 PRINT USING "###";AS%
7080 PRINT " TOTAL REMAINING: ";
7090 PRINT USING "###";AL%
7100 PRINT :PRINT :PRINT :PRINT
7110 INPUT "ENTER PERCENTAGE FOR FIELD: ";PCT%(I)
7120 IF PCT%(I)>AL% OR PCT%(I)=0 THEN PCT%(I)=AL%
7130 AL%=AL%-PCT%(I)
7140 AS%=AS%+PCT%(I)
7150 PRINT :PRINT
7160 INPUT "ENTER DESCRIPTION OF FIELD: ";PXT$(I)
7170 IF LEN(PXT$(I))>15 THEN PRINT "FIELD TOO LONG -
      15 CHAR. MAX": GOTO 7160
7180 IF AL%=0 GOTO 7200
7185 I=I+1
7190 GOTO 7050
7200 NP%=I
7210 IF NP%=1 THEN 7030
```



722Ø CLS  
723Ø RETURN



**Appendix E**

## **Osborne, Kaypro and Other CP/M Computers**

All that you need to connect Delta to an Osborne 1 or Kaypro computer is a cable. Your Delta dealer can provide the cable that you need.

### ***Setting the switches***

When connecting Delta to an Osborne 1, Kaypro, or other CP/M computer, we recommend that you set the DIP switches in Delta as shown below. (Although our chart indicates switch 2-2 set for a parallel interface, a serial interface will work also.)

#### ***Recommended DIP Switch Settings for Osborne 1***

| <b>Switch</b> | <b>Setting</b> | <b>Function</b>           |
|---------------|----------------|---------------------------|
| 1-1           | ON             | 11 inch page size         |
| 1-2           | ON             | Normal print density      |
| 1-3           | ON             | 10 CPI pitch              |
| 1-4           | ON             | Normal characters         |
| 1-5           | ON             | 1/6 inch line feed        |
| 1-6           | ON             | U.S.A. Character set      |
| 1-7           | ON             |                           |
| 1-8           | ON             |                           |
| 2-1           | ON             | Paper-out detector active |
| 2-2           | OFF            | Parallel interface        |
| 2-3           | OFF            | 8-bit interface           |
| 2-4           | OFF            | No auto line feed         |

When you connect your printer to your Osborne 1 you must use the SETUP program to tell the computer whether Delta is connected to the Osborne 1's serial or parallel interface (either will work).



**Osborne 1 Parallel Cable**

| Delta   |          | Osborne 1 |          |
|---------|----------|-----------|----------|
| Pin No. | Function | Pin No.   | Function |
| 2       | DATA1    | 1         | DATA0    |
| 6       | DATA5    | 2         | DATA4    |
| 3       | DATA2    | 3         | DATA1    |
| 7       | DATA6    | 4         | DATA5    |
| 4       | DATA3    | 5         | DATA2    |
| 8       | DATA7    | 6         | DATA6    |
| 5       | DATA4    | 7         | DATA3    |
| 9       | DATA8    | 8         | DATA7    |
| 1       | STROBE   | 11        | STROBE   |
| 11      | BUSY     | 15        | BUSY     |
| 16      | SIG GND  | 16        | SIG GND  |

**Kaypro Parallel Cable**

| Delta   |          | Kaypro  |          |
|---------|----------|---------|----------|
| Pin No. | Function | Pin No. | Function |
| 1       | STROBE   | 1       | STROBE   |
| 2       | DATA1    | 2       | DATA1    |
| 3       | DATA2    | 3       | DATA2    |
| 4       | DATA3    | 4       | DATA2    |
| 5       | DATA4    | 5       | DATA2    |
| 6       | DATA5    | 6       | DATA2    |
| 7       | DATA6    | 7       | DATA2    |
| 8       | DATA7    | 8       | DATA2    |
| 9       | DATA8    | 9       | DATA8    |
| 11      | BUSY     | 11      | BUSY     |
| 16      | SIG GND  | 16      | SIG GND  |

**Using MBASIC**

Many CP/M computers use Microsoft BASIC (called MBASIC). If you have a CP/M-80 computer that uses Microsoft BASIC the program listings given here should work with your computer also.

MBASIC is a very close relative of the IBM-Microsoft BASIC used in this book. The only difference is that MBASIC "interprets" CHR\$(9) and substitutes a group of spaces to simulate a tab. You can send a horizontal tab to Delta by using CHR\$(137) instead of CHR\$(9).

Microsoft BASIC uses the "L" prefix on several commands to



direct them to the printer. To list programs on the printer, just type LLIST. To direct program output to the printer, use LPRINT in place of PRINT.

Some versions of Microsoft BASIC will add a carriage return and line feed at the end of every 80 (or sometimes 132) characters. To print lines longer than 80 (or 132) characters (as when doing dot graphics) you must define a wider printer width. The following statement will prevent the computer from inserting unwanted codes.

```
10 WIDTH LPRINT 255
```

### **Program listings**

The following programs are in Microsoft BASIC for the Osborne 1.

#### **Chart program**

```
100 WIDTH LPRINT 255
110 GOSUB 1000
120 GOSUB 2000
130 LPRINT "*REGULAR*"
140 GOSUB 3000
150 LPRINT "*DOUBLE STRIKE*"
160 LPRINT DOUBLE.STRIKE$;
170 GOSUB 3000
180 LPRINT "*EMPHASIZED*"
190 EMPHASIZED=TRUE
200 GOSUB 3000
210 LPRINT "*DOUBLE STRIKE AND EMPHASIZED*"
220 LPRINT DOUBLE.STRIKE$ EMPHASIZED$;
230 GOSUB 3000
240 END
1000 REM
1060 ITALIC$=CHR$(27) + CHR$(52)
1070 ROMAN$=CHR$(27) + CHR$(53)
1090 ENLARGED$=CHR$(27) +CHR$(87)+CHR$(1)
1100 NOT.ENLARGED$=CHR$(27)+CHR$(87)+CHR$(0)
1110 PICA$=CHR$(27)+CHR$(66)+CHR$(1)
1120 ELITE$=CHR$(27)+CHR$(66)+CHR$(2)
1130 CONDENSED$=CHR$(27)+CHR$(66)+CHR$(3)
1150 EMPHASIZED$=CHR$(27)+CHR$(69)
1160 NOT.EMPHASIZED$=CHR$(27)+CHR$(70)
1170 DOUBLE.STRIKE$=CHR$(27)+CHR$(71)
```



```
1180 NOT.DOUBLE.STRIKE$=CHR$(27)+CHR$(72)
1190 UNDERLINED$=CHR$(27)+CHR$(45)+CHR$(1)
1200 NOT.UNDERLINED$=CHR$(27)+CHR$(45)+CHR$(0)
1210 SUPERScript$=CHR$(27)+CHR$(83)+CHR$(0)
1220 SUBSCRIPT$=CHR$(27)+CHR$(83)+CHR$(1)
1230 NOT.SCRIPTED$=CHR$(27)+CHR$(84)
1240 RESET.ALL$=NOT.EMPHASIZED$+NOT.UNDERLINED$+
    NOT.DOUBLE.STRIKE$
1250 RESET.ALL$=RESET.ALL$+ROMAN$+PICA$+
    NOT.ENLARGED$
1270 TRUE=1: FALSE=0
1280 REGULAR.HEADINGS$=STRING$(27,"*")+ "REGULAR"+
    STRING$(27,"*")
1290 RETURN
2000 REM
2050 LPRINT RESET.ALL$
2060 LPRINT ENLARGED$ "    NORMAL    ENLARGED"
2070 LPRINT RESET.ALL$;
2080 LPRINT UNDERLINED$;
2090 LPRINT CONDENSED$ "CONDENSED ";
2100 LPRINT ELITE$ "    ELITE    ";
2110 LPRINT PICA$ "    PICA    ";
2120 LPRINT CONDENSED$ "CONDENSED ";
2130 LPRINT ELITE$ "    ELITE    ";
2140 LPRINT PICA$ "    PICA    ";
2150 LPRINT RESET.ALL$
2160 RETURN
3000 REM
3050 ITALICS=FALSE:LPRINT ROMAN$;
3060 UNDERLINED=FALSE:LPRINT NOT.UNDERLINED$;
3070 ENLARGED=FALSE:LPRINT NOT.ENLARGED$;
3080 PICA=FALSE
3100 LPRINT CONDENSED$;
3110 GOSUB 3500
3120 LPRINT ELITE$;
3130 GOSUB 3500
3140 LPRINT PICA$;:PICA=TRUE
3150 GOSUB 3500
3170 IF ENLARGED=TRUE THEN LPRINT:GOTO 3190
3180 LPRINT ENLARGED$;:ENLARGED=TRUE:GOTO 3080
3190 IF UNDERLINED=TRUE THEN LPRINT:GOTO 3210
3200 LPRINT UNDERLINED$;:UNDERLINED=TRUE:GOTO 3070
3210 IF ITALICS=TRUE THEN LPRINT RESET.ALL$:RETURN
3220 LPRINT ITALIC$;:ITALICS=TRUE:GOTO 3060
3500 REM
```



```
3550 BLANK$=STRING$(6,32):FOUR.DOT$="...."
3560 IF EMPHASIZED=FALSE THEN LPRINT"ABcd";GOTO
    3610
3570 IF PICA=FALSE THEN LPRINT FOUR.DOT$;:GOTO 3590
3580 LPRINT EMPHASIZED$ "ABcd ";
3590 IF ENLARGED=TRUE THEN LPRINT " ";:ELSE LPRINT
    BLANK$;
3600 RETURN
3610 REM
3620 IF ENLARGED=TRUE THEN LPRINT " ";:RETURN
3630 LPRINT SUPERScript$; "Xx";
3640 LPRINT SUBScript$; "Yy ";
3650 LPRINT NOT.SCRIPTED$;
3660 RETURN
```

### Special character chart program

```
10 FOR J=160 TO 255 STEP 8
20 FOR I=J TO J+7
30 LPRINT I "= "CHR$(I);" ";
40 NEXT:LPRINT:NEXT
```

### Macro program

```
10 LPRINT CHR$(27)+" ";
20 LPRINT CHR$(18);
30 LPRINT CHR$(27)"W"CHR$(0);
40 LPRINT CHR$(27)"F";
50 LPRINT CHR$(27)"H";
60 LPRINT CHR$(27) "-" CHR$(0);
70 LPRINT CHR$(27) "T";
80 LPRINT CHR$(27) "5";
90 LPRINT CHR$(30);
```

### Bridge hand program

```
10 WIDTH LPRINT 255
20 GOSUB 1000
30 GOSUB 2000
40 GOSUB 3000
50 GOSUB 4000
60 END
1000 REM
1020 DIM HAND(4),DECK(52),CARD$(13),SUIT$(3)
1030 CARD$(1)=" 2":CARD$(2)=" 3":CARD$(3)="
    4":CARD$(4)=" 5":CARD$(5)=" 6"
```



```
1040 CARD$(6)=" 7":CARD$(7)=" 8":CARD$(8)="
    9":CARD$(9)=" 10"
1050 CARD$(10)=" J":CARD$(11)=" Q":CARD$(12)="
    K":CARD$(13)=" A"
1060 SUIT$(0)="S":SUIT$(1)="H":SUIT$(2)="D":
    SUIT$(3)="C"
1070 INPUT "Random number seed ";I
1080RANDOMIZE I
1090 RETURN
2000 REM
2010 LPRINT CHR$(27)CHR$(68)CHR$(20)CHR$(40)CHR$(0)
2020 LPRINT CHR$(27)CHR$(43)CHR$(27)CHR$(36)CHR$(0)
    CHR$(27)CHR$(69)CHR$(30)
2030 LPRINT CHR$(27)CHR$(42)CHR$(0)
2040 FOR I=1 TO 4
2050 LPRINT CHR$(27)CHR$(42)CHR$(1);
2060 FOR J=1 TO 13
2070 READ X:LPRINT CHR$(X);
2080 NEXT J
2090 NEXT I
2100 LPRINT
2110 RETURN
2120 DATA 72,11,4,10,20,10,52,72,52,10,20,10,4
2130 DATA 83,11,16,8,20,8,86,41,86,8,20,8,16
2140 DATA 67,11,8,16,8,18,65,62,65,18,8,16,8
2150 DATA 68,11,8,0,28,0,62,65,62,0,28,0,8
3000 REM
3010 FOR CARD=1 TO 52
3020 X=INT(RND*4+1)
3030 IF HAND(X)=13 THEN 3020 ELSE HAND(X)=HAND(X)+1
3040 DECK(CARD)=X
3050 NEXT CARD
3060 RETURN
4000 REM
4010 LPRINT CHR$(27) "!" SPC(20) "NORTH"
4020 LPRINT CHR$(27) "$" CHR$(1) CHR$(27)CHR$(70);
4030 HAND=1
4040 FOR SUIT=0 TO 3
4050 LPRINT SPC(20);
4060 GOSUB 4300
4070 LPRINT
4080 NEXT SUIT
4090 LPRINT CHR$(27) "!" "WEST" SPC(40) "EAST"
4100 LPRINT CHR$(27) "$" CHR$(1) CHR$(27)CHR$(70);
4110 FOR SUIT=0 TO 3
```



```
4120 HAND=2
4130 GOSUB 4300
4140 LPRINT TAB(45);
4150 HAND =3
4160 GOSUB 4300
4170 LPRINT
4180 NEXT SUIT
4190 LPRINT CHR$(27) "!" SPC(20) "SOUTH"
4200 LPRINT CHR$(27) "$" CHR$(1)CHR$(27)CHR$(70);
4210 HAND=4
4220 FOR SUIT=0 TO 3
4230 LPRINT SPC(20);
4240 GOSUB 4300
4250 LPRINT
4260 NEXT SUIT
4270 LPRINT CHR$(27) "$" CHR$(0)CHR$(27)CHR$(70)
4280 RETURN
4300 LPRINT SUIT$(SUIT);
4310 FOR CARD=13 TO 1 STEP -1
4320 IF DECK(SUIT*13+CARD)=HAND THEN LPRINT
      CARD$(CARD);
4330 NEXT CARD
4340 RETURN
```

### **Numeral program**

```
30 DEF.DOWN.CHAR$=CHR$(27)+CHR$(42)+CHR$(1)
40 DOWN.CHAR.PROP$=CHR$(27)+CHR$(88)+CHR$(1)
50 NOT.DOWN.CHAR.PROP$=CHR$(27)+CHR$(88)+CHR$(0)
60 LINE.7$=CHR$(27)+CHR$(49):LINE.12$=CHR$(27)+
  CHR$(50)
70 FOR N1=160 TO 200
80 LPRINT DEF.DOWN.CHAR$;
90 LPRINT CHR$(N1);
100 READ N2
110 LPRINT CHR$(N2);
120 FOR S=1 TO 11
130 READ MS
140 LPRINT CHR$(MS);
150 NEXT S
160 NEXT N1
180 ASCII=160
190 FOR NUM=0 TO 9
200 NUMERAL.TOP$(NUM)=CHR$(ASCII+0)+CHR$(ASCII+1)
210 NUMERAL.BOT$(NUM)=CHR$(ASCII+2)+CHR$(ASCII+3)
220 ASCII=ASCII+4
```



```
230 NEXT NUM
240 BLANK$=CHR$(200)
250 LPRINT DOWN.CHAR.PROP$;LINE.7$
260 FOR NUM=0 TO 9
270 LPRINT NUMERAL.TOP$(NUM);BLANK$;
280 NEXT NUM
290 LPRINT
300 FOR NUM=0 TO 9
310 LPRINT NUMERAL.BOT$(NUM);BLANK$;
320 NEXT NUM
330 LPRINT NOT.DOWN.CHAR.PROP$;LINE.12$
340 REM ZERO
350 DATA 11,0,96,16,104,16,44,30,14,0,2,1
360 DATA 11,2,1,2,1,6,8,38,88,32,88,32
370 DATA 11,3,12,19,12,51,0,96,0,96,0,96
380 DATA 11,0,32,0,48,0,28,3,12,3,4,3
390 REM ONE
400 DATA 11,0,0,0,0,0,4,0,4,0,4,126
410 DATA 10,12,114,12,114,12,2,0,0,0,0,0
420 DATA 11,64,0,64,0,64,0,64,32,80,47,80
430 DATA 10,47,80,47,64,0,64,0,64,0,0,0
440 REM TWO
450 DATA 11,0,0,0,0,0,12,16,14,0,6,0
460 DATA 11,3,0,3,0,70,56,70,56,4,24,0
470 DATA 11,64,0,64,32,64,32,80,32,80,40,64
480 DATA 11,44,64,38,65,34,65,32,80,32,88,0
490 REM THREE
500 DATA 11,0,0,0,0,0,0,4,2,4,2,4
510 DATA 11,34,84,34,92,34,76,34,68,2,64,0
520 DATA 11,16,0,48,0,56,64,48,64,32,64,32
530 DATA 11,64,32,64,48,10,54,10,22,10,6,1
540 REM FOUR
550 DATA 11,0,0,0,0,0,0,64,36,88,32,16
560 DATA 11,0,0,64,32,64,56,64,60,2,12,0
570 DATA 11,0,8,4,10,5,10,5,8,4,72,4
580 DATA 11,88,38,89,38,89,6,73,4,8,6,0
590 REM FIVE
600 DATA 11,0,0,0,0,64,32,84,50,76,34,68
610 DATA 10,34,68,34,68,34,68,2,68,2,0,0
620 DATA 10,0,32,24,101,24,97,0,64,0,64,0
630 DATA 11,64,0,96,1,48,15,48,15,16,15,0
640 REM SIX
650 DATA 11,0,96,0,112,0,120,0,92,0,102,0
660 DATA 11,98,0,98,0,98,0,70,0,14,0,6
670 DATA 11,7,8,23,8,55,8,99,0,65,0,64
```



```

680 DATA 11,0,96,0,112,1,62,1,30,1,14,0
690 REM SEVEN
700 DATA 11,0,16,8,6,8,6,8,6,8,6,8
710 DATA 10,70,8,102,8,54,8,6,0,2,0,0
720 DATA 11,0,64,0,96,0,120,0,124,0,30,1
730 DATA 10,6,1,0,0,0,0,0,0,0,0,0
740 REM EIGHT
750 DATA 11,0,0,0,0,24,36,24,102,24,102,0
760 DATA 11,67,0,67,0,99,28,34,28,34,28,0
770 DATA 11,12,18,44,19,108,19,96,1,64,0,64
780 DATA 11,0,96,1,112,15,48,15,16,14,0,0
790 REM NINE
800 DATA 11,0,0,120,4,120,6,120,6,0,3,0
810 DATA 11,3,0,3,0,67,4,123,4,122,4,120
820 DATA 11,48,0,56,0,113,0,99,0,99,0,99
830 DATA 11,0,115,0,57,0,31,0,15,0,7,0
840 REM SPACE
850 DATA 11,0,0,0,0,0,0,0,0,0,0,0,

```

### Download utility program

```

10 DIM Z(8,12),MM(11)
15 DEF FNLOCATES$(X,Y)=CHR$(27)+CHR$(61)+CHR$(X)+
    CHR$(Y+32)
20 PRINT CHR$(26);:GOSUB 660
30 CS$="<>":SC$="[]":BIT=0:SS$="00"
40 A$=INKEY$:IF A$="" THEN 40
50 IF A$="Q" THEN PRINT CHR$(26):END
60 IF A$="P" THEN GOSUB 680:GOTO 40
70 IF A$="E" THEN PRINT CHR$(26):GOSUB 90:GOSUB 260:
    GOTO 40
80 GOTO 40
90 REM EDIT MODE
95 H=1:G=1:X=1:Y=1
96 FOR I=1 TO 11:MM(I)=0:NEXT I
100 PRINT CHR$(26)
110 PRINT
120 PRINT "      M1 M2 M3 M4 M5 M6 M7 M8 M9 M10 M11"
130 FOR I=0 TO 7:PRINT "      ";:FOR J=1 TO 11:
140 PRINT "!=";:NEXT J:PRINT "!=":IF I<7 THEN PRINT
    2^I
150 NEXT I
160 PRINT FNLOCATES$(20,1);"R)IGHT L)EFT U)P D)OWN
    C)LEAR Q)UIT"
180 RETURN
260 REM **** SINGLE CHARACTER INPUT @ EDIT LEVEL****

```



```

270 PRINT FNLOCATE$(4,7);:PRINT CS$;:PRINT
  FNLOCATE$(20,40);
280 A$=INKEY$: IF A$="" THEN 280
300 IF A$="L" THEN GOSUB 390:GOTO 370
310 IF A$="R" THEN GOSUB 410:GOTO 370
320 IF A$="D" THEN GOSUB 430:GOTO 370
330 IF A$="U" THEN GOSUB 450:GOTO 370
340 IF A$="I" THEN GOSUB 470:GOTO 370
350 IF A$="C" THEN GOSUB 490:GOTO 370
360 IF A$="Q" THEN GOSUB 500:GOTO 380
370 GOTO 280
380 RETURN: REM ****END OF INPUT****
390 GOSUB 920:Y=Y-3:H=H-1:IF Y<1 THEN Y=1:H=1
400 GOSUB 950 : RETURN
410 GOSUB 920:Y=Y+3:H=H+1:IF Y>31 THEN Y=31:H=11
420 GOSUB 950:RETURN
430 GOSUB 920:X=X+2:G=G+1:IF X>13 THEN X=13:G=7
440 GOSUB 950:RETURN
450 GOSUB 920:X=X-2:G=G-1:IF X<1 THEN X=1:G=1
460 GOSUB 950:RETURN
470 IF Z(G,H-1)=1 OR Z(G,H+1)=1 THEN RETURN
480 Z(G,H)=1:PRINT FNLOCATE$(X+3,Y+6); SS$;:RETURN
490 Z(G,H)=0:PRINT FNLOCATE$(X+3,Y+6); CS$;:RETURN
500 REM ****GET OUT OF EDIT MODE****
520 IF Z(G,H)=1 THEN PRINT FNLOCATE$(X+3,Y+6);
  SC$;:GOTO 540
530 IF Z(G,H)=0 THEN PRINT FNLOCATE$(X+3,Y+6); " ";
540 REM **** PRINT THE COLUMN - VALUES****
550 FOR I=1 TO 11: FOR J=1 TO 7
560 MM(I)=MM(I)+Z(J,I)*2^(J-1):NEXT J:NEXT I
570 J=0:FOR I=1 TO 11:PRINT FNLOCATE$(19,6+J);
  RIGHT$(STR$(MM(I)),3);:J=J+3:NEXT I
572 PRINT
575 PRINT FNLOCATE$(20,1);STRING$(45," ");
580 GOSUB 660:RETURN
660 FOR I=1 TO 7:FOR J=1 TO 11:Z(I,J)=0:NEXT J:NEXT
  I
670 PRINT FNLOCATE$(22,2);:PRINT "E) EDIT      P)
  PRINTER      Q) QUIT ";:RETURN
680 REM ****PRINT MODE****
690 PRINT FNLOCATE$(20,1);:INPUT "NORMAL OR
  PROPORTIONAL      (N/P) ";AN$
700 IF AN$="N" THEN PR=0:GOTO 750
710 IF AN$="P" THEN GOTO 730
720 GOTO 690

```



```

730 GOSUB 2000:PRINT FNLOCATE$(20,1);:INPUT "THE
    PROPORTIONAL DATA (4-11) "; PR
740 IF PR<4 OR PR>11 THEN 730
750 GOSUB 2000:PRINT FNLOCATE$(20,1);:INPUT "IF
    SHIFTED ENTER 1 ELSE ENTER 0 ";SH
760 IF SH<0 OR SH>1 THEN GOTO 750
770 GOSUB 2000:PRINT FNLOCATE$(20,1);:INPUT "ASCII
    CODE (33-126 OR 160-254) ";AS
780 IF (AS<33 OR AS>126) AND (AS<160 OR AS>254) THEN
    770
785 PRINT
790 PRINT FNLOCATE$(20,1);STRING$(50," ")
800 IF SH=1 THEN SH=16 ELSE SH=0
810 N1=AS:N2=PR+SH
820 FOR I=1 TO 11:MM$=MM$+CHR$(MM(I)):NEXT I
830 LPRINT
    CHR$(27);"*";CHR$(1);CHR$(N1);CHR$(N2);MM$
840 IF AN$="N" THEN LPRINT CHR$(27);"$";CHR$(1):GOTO
    860
850 LPRINT CHR$(27);"X";CHR$(1)
860 FOR I=1 TO 20:LPRINT CHR$(N1);" ";:NEXT I:LPRINT
870 LPRINT CHR$(14);:FOR I=1 TO 10:LPRINT CHR$(N1);"
    ";:NEXT I:LPRINT CHR$(20)
880 LPRINT CHR$(15);:FOR I=1 TO 20:LPRINT CHR$(N1);"
    ";:NEXT I:LPRINT CHR$(18)
890 IF AN$="N" THEN LPRINT CHR$(27);"$";CHR$(0):GOTO
    910
900 LPRINT CHR$(27);"X";CHR$(0)
910 LPRINT CHR$(27);"@":MM$="":RETURN:REM ****END OF
    PRINT MODE****
920 IF Z(G,H)=0 THEN PRINT FNLOCATE$(X+3,Y+6); " ";
930 IF Z(G,H)=1 THEN PRINT FNLOCATE$(X+3,Y+6); SC$;
940 RETURN
950 IF Z(G,H)=1 THEN PRINT FNLOCATE$(X+3,Y+6); SS$;
960 IF Z(G,H)=0 THEN PRINT FNLOCATE$(X+3,Y+6); CS$;
970 RETURN
2000 PRINT FNLOCATE$(20,1);STRING$(50," ")
2010 RETURN

```

### Delta Plot program

```

10 PRINT CHR$(26)
20 PRINT "":PRINT "":PRINT ""
30 PRINT "THIS PROGRAM TAKES ABOUT TWO"
40 PRINT "MINUTES TO RUN, PLEASE TURN"
50 PRINT "ON YOUR PRINTER AND STAND BY"

```



```
100 REM DELTA-PLOT
110 DIM BIT%(76,14)
1000 REM SET PROGRAM CONSTANTS
1010 MASK%(1)=128:MASK%(4)=16
1020 MASK%(2)=64 :MASK%(5)=8
1030 MASK%(3)=32 :MASK%(6)=4
1040 LX=20      :LY=20
1050 XFAC=72/LX :YFAC=87/LY
2000 REM PLOT CURVE
2010 RAD=9
2020 X1=19 :Y1=10
2030 FOR ANG%=0 TO 360 STEP 10
2040 R1=ANG%*6.28/360
2050 X2=RAD*COS(R1)+10 :Y2=RAD*SIN(R1)+10
2060 GOSUB 4000
2070 NEXT ANG%
3000 REM SEND BIT IMAGE MAP TO PRINTER
3010 LPRINT CHR$(27);"A";CHR$(6)
3020 FOR ROW%=0 TO 14
3030 A$=""
3040 LPRINT CHR$(27);"K";CHR$(75);CHR$(0);
3050 FOR COL% = 1 to 75
3060 LPRINT CHR$(BIT%(COL%,ROW%));
3070 NEXT COL%
3080 LPRINT
3090 NEXT ROW%
3100 LPRINT CHR$(27);"A";CHR$(12)
3110 END
4000 REM DRAW A LINE FROM X1,Y1 TO X2,Y2
4010 XL=X2-X1 :YL=Y2-Y1
4020 NX=ABS(XL*XFAC) : NY=ABS(YL*YFAC)
4030 IF NX<NY THEN NX=NY
4040 NS%=INT(NX+1)
4050 DX=XL/NS% : DY=YL/NS%
4060 FOR I%=1 TO NS%
4070 X1=X1+DX : Y1=Y1+DY
4080 GOSUB 5000
4090 NEXT I%
4100 RETURN
5000 REM PLOT A POINT AT X1,Y1
5010 XX=X1*XFAC :YY=Y1*YFAC
5020 COL%=INT(XX)+1
5030 ROW%=INT(YY/6)
5040 XIT%=INT(YY-ROW%*6)+1
```



```
5050 BIT%(COL%,ROW%)=BIT%(COL%,ROW%) OR MASK%(XIT%)
5060 RETURN
```

### Pie chart program

```
110 ESC$=CHR$(27):LF$=CHR$(10)
120 FF$=CHR$(12):VTAB$=CHR$(11)
130 EMPHASIZED$=ESC$="E":NOT.EMPHASIZED$=ESC$+"F"
1020 DIM BIT%(190,36),A$(36),PCT%(25),TEXT$(42),
    PIECETEXT$(25)
1030 MASK%(1)=128:MASK%(4)=16
1040 MASK%(2)=64 :MASK%(5)=8
1050 MASK%(3)=32 :MASK%(6)=4
1060 LX=20:LY=20
1070 LXFAC=190/LX:LYFAC=216/LY
1080 FOR I=1 TO 42
1090 TEXT$(I)=SPACE$(80)
1100 NEXT I
1110 GOSUB 7000
1120 PRINT CHR$(26)
1130 PRINT "THIS PROGRAM TAKES ABOUT FIVE MINUTES"
1140 PRINT "PLEASE HAVE YOUR PRINTER READY AND "
1150 PRINT "STAND BY "
2010 RAD=9
2020 X1=19:Y1=10
2030 PRINT " ";
2040 FOR ANG%=0 TO 360 STEP 5
2050 RANG=ANG%*6.28/360
2060 X2=RAD*COS(RANG)+10:Y2=RAD*SIN(RANG)+10
2070 GOSUB 4000
2080 NEXT ANG%
2090 FOR PIECE%=1 TO NUMBER.PIECES%
2100 X1=10:Y1=10
2110 TOTAL.PCT%=TOTAL.PCT%+PCT%(PIECE%)
2120 ANG%=360*TOTAL.PCT%*.01
2130 RANG=ANG%*6.28/360
2140 X2=RAD*COS(RANG)+10:Y2=RAD*SIN(RANG)+10
2150 GOSUB 4000
2160 GOSUB 6000
2170 NEXT PIECE%
3010 PRINT
3020 FOR ROW%=0 TO 35
3030 A$(ROW%)=""
3040 FOR COL%=1 TO 190
3050 A$(ROW%)=A$(ROW%)+CHR$(BIT%(COL%,ROW%))
3060 NEXT COL%
```



```
3070 PRINT CHR$(176);CHR$(176);
3080 NEXT ROW%
3090 PRINT
3091 LPRINT SPACE$(40-LEN(TITLE$)/
2);EMPHASIZED$;TITLE$;NOT. EMPHASIZED$;LF$
3100 LPRINT VTAB$;VTAB$;VTAB$
3110 LPRINT ESC$;"A";CHR$(3)
3120 LPRINT TEXT$(1);LF$;TEXT$(2);LF$;TEXT$(3):LF$
3130 FOR ROW%=0 TO 35
3140 LPRINT "
"ESC$;"K";CHR$(190)CHR$(0);
3150 LPRINT A$(ROW%)
3160 LPRINT TEXT$(ROW%+4)
3170 LPRINT CHR$(176);CHR$(176);
3180 NEXT ROW%
3190 LPRINT TEXT$(40);LF$
3200 LPRINT TEXT$(41);LF$
3210 LPRINT TEXT$(42);LF$
3220 LPRINT ESC$;"2";FF$
3230 END
4000 '
4010 XL=X2-X1:YL=Y2-Y1
4020 NX=ABS(XL*LXFAC):NY=ABS(YL*LYFAC)
4030 IF NX<NY THEN NX=NY
4040 NS%=INT(NX+1)
4050 DX=XL/NS%:DY=YL/NS%
4060 FOR I%=1 TO NS%
4070 X1=X1+DX:Y1=Y1+DY
4080 GOSUB 5000
4090 NEXT I%
4100 PRINT CHR$(29);CHR$(205);CHR$(175);
4110 RETURN
5000 '
5010 XX=X1*LXFAC:YY=Y1*LYFAC
5020 COL%=INT(XX)+1
5030 ROW%=INT(YY/6)
5040 XIT%=INT(YY-ROW%*6)+1
5050 BIT%(COL%,ROW%)=BIT%(COL%,ROW%) OR MASK%(XIT%)
5060 RETURN
6000 REM
6010 MIDANG%=(ANG%+PREVANG%)/2
6020 RANG=MIDANG%*6.28/360
6030 X3=INT(20*SIN(RANG)):Y3=INT(22*COS(RANG))
6040 X4=22+X3:Y4=40+Y3
```



```
6050 IF MIDANG%>270 OR MIDANG%<90 THEN
    MID$(TEXT$(X4),Y4)=PIECETEXT$(PIECE%) ELSE
    MID$(TEXT$(X4),Y4-LEN(PIECETEXT$(PIECE%)))
    =PIECETEXT$(PIECE%)
6060 PREVANG%=ANG%
6070 RETURN
7000 '
7010 PRINT CHR$(26):PRINT:PRINT:PRINT
7020 INPUT"ENTER TITLE FOR CHART: ";TITLE$
7030 AMT.SOFAR%=0;AMT.LEFT%=100
7040 FOR I=1 TO 24
7050 PRINT CHR$(26);"                ENTER PARAMETERS FOR
    PIE CHART"
7060 PRINT                "                TOTAL SO FAR";
7070 PRINT AMT.SOFAR%
7080 PRINT                "                TOTAL REMAINING";
7090 PRINT AMT.LEFT%
7100 PRINT:PRINT:PRINT:PRINT
7110 INPUT "ENTER PERCENTAGE FOR FIELD: ";PCT%(I)
7120 IF PCT%(I)>AMT.LEFT% OR PCT%(I)=0 THEN
    PCT%(I)=AMT.LEFT%
7130 AMT.LEFT%=AMT.LEFT%-PCT%(I)
7140 AMT.SOFAR%=AMT.SOFAR%+PCT%(I)
7150 PRINT:PRINT
7160 INPUT "ENTER DESCRIPTION OF FIELD:
    ";PIECETEXT$(I)
7170 IF LEN(PIECETEXT$(I))>15 THEN PRINT "FIELD TOO
    LONG - 15 CHAR. MAX":GOTO 7160
7180 IF AMT.LEFT%=0 THEN 7200
7190 NEXT I
7200 NUMBER.PIECES%=I
7210 IF NUMBER.PIECES%=0 THEN 7030
7220 PRINT CHR$(26)
7230 RETURN
```



